
Motivation

I chose this topic because of my curiosity about the extent to which software

verification tools can be used not only for proving properties of programs, but also

for reasoning about highly abstract mathematical systems. My interest in this direc-

tion began during the course “Verification Driven Program Development”, in the first

semester of my final year. As part of that course, I worked in a team of three students

on a project that formalized natural deduction for propositional logic in Dafny and

proved its soundness and completeness. The project later led to an article accepted at

the ECOOP VeriLang’26 workshop, which encouraged me to continue exploring this

topic in greater depth.

Software verification is an important field of computer science because it aims to

provide strong guarantees about the correctness of systems. In many areas, especially

those involving critical software, informal testing is not sufficient to exclude subtle

errors. Formal verification offers a more rigorous approach by using mathematical

specifications and machine-checked proofs. While Dafny is often used for verifying

programs and data structures, I was especially interested in exploring the reach of its

capabilities by using the formalization and verification of natural deduction and its

meta-properties as a case study.

Verifying logic systems is important because such systems form the foundation

of programming languages, theorem provers, proof assistants, and other tools that

rely on formal reasoning. Establishing properties such as soundness and completeness

increases confidence that the rules of a proof system behave as intended.

A secondary motivation for this thesis was the role of AI models in assisting proof

development. Since public verification code is relatively scarce, it is important to study

and submit findings on how such models perform in this setting. Therefore, this the-

sis presents a case study focused on natural deduction, Dafny formalization, and the

interaction between automated verification and AI-assisted development.

Contributions

The main contributions of this thesis consist of Dafny-verified formalizations of

natural deduction and an experimental study of AI-assisted development for formal

verification. More specifically, the contributions are the following:

• Contributions regarding Dafny formalizations:

– A formalization of natural deduction using a height-bounded judgment, to-

gether with machine-checked proofs of soundness and completeness;

– A formalization of natural deduction using derivation-tree data structures,

together with machine-checked proofs of soundness and completeness;

– A formalization of natural deduction using least predicates, together with

machine-checked proofs of soundness and completeness.

• Contributions regarding LLM-assisted verification development:

– Observations on the capabilities and limitations of chat-based large language

models in assisting the development of the above formalizations, with par-

ticular attention to Claude Sonnet 4.5 and OpenAI Codex 5.2 and 5.4;

– An experimental assessment of agentic LLM performance in adapting proofs

from one formalization approach to another;

– An experimental assessment of agentic LLM performance in constructing

verification proofs from scratch.

Introduction

Dafny is an auto-active verification language: the user writes programs, specifica-

tions, and proof annotations, while the verification conditions are discharged automat-

ically by the underlying toolchain. Dafny translates annotated programs into Boogie,

and Boogie relies on an SMT solver, usually Z3, to prove the generated obligations.

This makes Dafny suitable not only for verifying ordinary programs and data struc-

tures, but also for developing machine-checked proofs about mathematical systems.

This thesis studies that broader use of Dafny through a case study in proposi-

tional logic. The object of the formalization is natural deduction, a proof system whose

rules define when a formula can be derived from a set of assumptions. The main

metatheoretical properties considered are soundness and completeness. Soundness

states that every syntactically derivable formula is semantically valid with respect to

its assumptions: if Γ ⊢ φ, then Γ |= φ. Completeness states the converse: if Γ |= φ, then

Γ ⊢ φ. Together, these properties establish the correspondence between derivability

and semantic consequence.

The mathematical results themselves are classical. The purpose of the thesis is

therefore not to prove a new theorem in logic, but to investigate how these standard

results can be represented and verified in Dafny. The development follows the theo-

retical approach presented in Carmen Bruni’s CS Waterloo material on soundness and

completeness for natural deduction, but the informal argument has to be transformed

into explicit definitions, lemmas, induction principles, and auxiliary data-structure rea-

soning that Dafny can verify.

The central research problem is to compare different Dafny representations of

derivability for propositional natural deduction. The thesis presents three verified for-

malizations. The first uses a height-bounded judgment, where derivability is indexed

by a proof-height bound. The second uses explicit derivation trees, where proof ob-

jects are represented by a datatype whose constructors correspond to natural deduc-

tion rules. The third uses Dafny’s least predicate mechanism, where derivability

is encoded directly as the least relation closed under the inference rules. Each for-

malization proves both soundness and completeness, but each one imposes different

proof-engineering costs.

A second theme of the thesis is the use of large language models in formal verifi-

cation development. Public verification code is less abundant than ordinary program-

ming code, and Dafny proofs are sensitive to small changes in specifications, lemma

statements, and induction structure. For this reason, the thesis records how generative

AI tools assisted the development of the formalizations. The observations focus on

tasks such as drafting proof skeletons, proposing auxiliary lemmas, completing repet-

itive cases, adapting proof ideas between representations, and identifying the points

where human proof design remained necessary.

The thesis is organized as follows. The chapter Formalization of Propositional

Logic introduces the common logical infrastructure used by the later developments.

It defines the datatype of propositional formulae, truth assignments, semantic evalu-

ation, semantic consequence, and formula validity. These definitions provide the se-

mantic side of the soundness and completeness statements.

The chapter Formalizing Natural Deduction using a Height-Bounded Judge-

ment presents the first natural deduction formalization. In this version, the main deriv-

ability predicate includes a numerical height argument, which makes recursive defi-

nitions acceptable to Dafny’s termination checker and supports induction over proof

height. The chapter defines the proof rules, proves soundness by induction on the

height bound, and proves completeness by formalizing the required semantic deduc-

tion and tautology arguments.

The chapter Findings in LLM-assisted Program Verification Development sum-

marizes the practical experience of using generative AI during the Dafny develop-

ment. It discusses where the assistant was useful, such as producing proof skeletons

and routine helper lemmas, and where it was unreliable, such as maintaining complex

invariants, avoiding incomplete lemmas, and preserving the intended specifications.

The chapter Derivation Tree Natural Deduction presents the second formaliza-

tion. Here, proofs are explicit objects: a derivation tree stores both the proof structure

and the assumptions and formula at each node. This representation is close to the

mathematical picture of proof trees, but it also requires additional infrastructure for

validity predicates, root projections, tree transformations, and weakening. The chap-

ter compares this proof-object approach with the height-bounded one and reports how

Codex was used to adapt and complete parts of the proof development.

The chapter Least-Predicate Natural Deduction presents the third formalization.

In this version, Dafny’s least-predicate mechanism represents derivability directly as

an inductively generated relation. This removes the need for explicit height indices

or separate derivation-tree validity checks. The chapter studies whether this represen-

tation gives a more natural and compact encoding of natural deduction, and it again

establishes soundness and completeness for the resulting proof relation.

The concluding chapter compares the three formalization styles and summarizes

the findings on LLM-assisted verification. The comparison highlights the trade-offs

between termination-oriented encodings, explicit proof objects, and native inductive

predicates in Dafny formalizations of logic and AI-supported proof engineering.

Chapter 1

Formalization of Propositional Logic

Propositional Logic is comprised of two components: the object language and the

semantics. The object language is represented in propositional formulae. The seman-

tics of proposition logic which are relevant in the meta-theory proof are truth assign-

ments, logical entailment (consequence), and formula validity.

The implementation of Propositional Logic is the same accross the formalizations.

In the formalization using Least Predicates, the function and predicates names may be

changed for some of the functions and predicates, the logic remains identical.

The determination logic for formula equivalence, formula satisfiability and for-

mula tree height are also present in the development, but they did not end up being

used in the proof of soundness and completeness.

1.1 Data structures

Propositional formulae are modelled using the algebraic datatype Formula, shown

in Listing 1.1 The datatype directly encodes the syntax of the object language, with con-

structors for propositional variables, negation, conjunction, disjunction, implication,

and the constants ⊥ and ⊤. Variables are represented by their names as strings. Most

propositional connectives are included, even though some could be defined in terms

of others.

Listing 1.1: Propositional Formulas Inductive Datatype
1 datatype Formula = Var(name: string)
2 | Not(f: Formula)
3 | And(f1: Formula, f2: Formula)
4 | Or(f1: Formula, f2: Formula)
5 | Implies(f1: Formula, f2: Formula)

6 | Bottom
7 | Top

Truth assignments are defined as total functions from variable names to Boolean

values, as shown in Listing 1.2.

Listing 1.2: Truth Assignment Type
1 type TruthAssignment = string -> bool

The recursive function value reveals the truth value of a formula given a truth

assignment. This is shown in Listing 1.3.

Listing 1.3: Truth Evaluation Function
1 function value(f: Formula, tau: TruthAssignment) : bool
2 decreases f
3 {
4 match f
5 case Var(x)
6 => tau(x)
7 case Not(p)
8 => !value(p, tau)
9 case And(p, q)

10 => value(p, tau) && value(q, tau)
11 case Or(p, q)
12 => value(p, tau) || value(q, tau)
13 case Implies(p, q)
14 => !value(p, tau) || value(q, tau)
15 case Bottom => false
16 case Top => true
17 }

Logical consequence is defined as the ghost predicate conseq, shown in List-

ing 1.4. Ghost predicates enable the expression of specification-only properties, typ-

ically involving quantifiers. These predicates are intended for verification rather than

execution. Since ghost predicates exist only at verification time and are erased in the

executable program, they permit the use of unbounded universal quantifiers over do-

mains that need not be finitely enumerable at runtime. Such quantifiers can be handled

logically by the verifier, even though compiling them into executable code is generally

not possible.

Listing 1.4: Logical Consequence Predicate
1 ghost predicate conseq(gamma : set<Formula>, psi : Formula)
2 {
3 forall tau ::
4 (forall phi :: phi in gamma
5 ==> value(phi, tau))

6 ==> value(psi, tau)
7 }

The validity of a formula ϕ says that: given any truth assignment the truth value

of the formula is positive. This is shown in Listing 1.5

Listing 1.5: Logical Consequence Predicate
1 ghost predicate valid(f : Formula)
2 {
3 forall tau :: value(f, tau)
4 }

Chapter 2

Formalizing Natural Deduction using a

Height-Bounded Judgement

This chapter presents the first Dafny formalization developed in this thesis: a for-

malization of propositional natural deduction based on a height-bounded derivability

judgment. The main idea is to represent derivability not by storing proof objects ex-

plicitly, but by defining a ghost predicate that states that a formula has a derivation

from a context whose height is bounded by a natural number. This additional height

parameter makes the recursive definition acceptable to Dafny’s termination checker

and gives a direct way to perform induction over derivations.

The formalization follows the standard mathematical presentation of natural de-

duction for propositional logic. Formulae are represented by an algebraic datatype,

truth assignments are represented as functions from variable names to Boolean val-

ues, and semantic consequence is expressed as a ghost predicate over all truth assign-

ments. On the syntactic side, each inference rule is encoded as a predicate, and the

main judgment NDMaxHeight is defined as a disjunction over these rule predicates. Or-

dinary provability, written as NDProof, is then recovered by existentially quantifying

over the proof height.

2.1 Natural Deduction

The inference rules of natural deduction are encoded as ghost predicates that

specify derivability. Because predicates in Dafny must provably terminate, an auxiliary

argument is used: proofHeight, a natural number that must decrease with every recur-

sive call. The height-bounded judgment is represented by NDMaxHeight(Γ, φ, h) (see

Listing 2.1). The predicate is true when there exists of a derivation of φ (phi : Formula,

the second argument) from Γ (gamma : set<Formula>, the first argument) whose height

is at most h (proofHeight : nat, the third argument).

Listing 2.1: Deduction by Tree of Derivation
1 ghost predicate NDMaxHeight(gamma : set<Formula>, phi : Formula, proofHeight: nat)
2 requires proofHeight >= 0
3 decreases proofHeight, 1
4 {
5 TopIntro(gamma, phi, proofHeight)
6 || Hypothesis(gamma, phi, proofHeight)
7 || NotElimination(gamma, phi, proofHeight)
8 || NotIntro(gamma, phi, proofHeight)
9 || OrElimination(gamma, phi, proofHeight)

10 || OrIntro1(gamma, phi, proofHeight)
11 || OrIntro2(gamma, phi, proofHeight)
12 || AndElimination1(gamma, phi, proofHeight)
13 || AndIntro(gamma, phi, proofHeight)
14 || AndElimination2(gamma, phi, proofHeight)
15 || ImpliesElimination(gamma, phi, proofHeight)
16 || ImpliesIntro(gamma, phi, proofHeight)
17 || BottomElimination(gamma, phi, proofHeight)
18 || NotNotElimination(gamma, phi, proofHeight)
19 }

The predicate NDMaxHeight is a disjuction over all inference rules in the proof sys-

tem. Some of the inference rules are illustrated in Listing 2.2. Note that these predicates

are mutually recursive, hence the decreases annotations use lexicographical ordering.

The predicate OrIntro1 encodes the standard introduction rule

Γ ⊢ φ1

Γ ⊢ φ1 ∨ φ2

,

and OrElimination encodes the case-analysis rule

Γ ⊢ φ1 ∨ φ2 Γ, φ1 ⊢ ψ Γ, φ2 ⊢ ψ
Γ ⊢ ψ

.

Listing 2.2: Deductive System Rule
1 ghost predicate OrIntro1(gamma : set<Formula>, phi : Formula, proofHeight : nat)
2 decreases proofHeight, 0
3 {
4 proofHeight > 0 &&
5 match phi
6 case Or(phi1, phi2) => NDMaxHeight(gamma, phi1, proofHeight - 1)
7 case _ => false
8 }
9 ...

10 ghost predicate OrElimination(gamma: set<Formula>, phi: Formula, proofHeight: nat)
11 decreases proofHeight, 0
12 {
13 proofHeight > 0 &&
14 exists phi1, phi2 ::
15 NDMaxHeight(gamma, Or(phi1, phi2), proofHeight - 1) &&
16 NDMaxHeight(gamma + {phi1}, phi, proofHeight - 1) &&
17 NDMaxHeight(gamma + {phi2}, phi, proofHeight - 1)
18 }

The use of a ghost-defined structure is necessary as some of the rules use ex-

istential quantifiers. Dafny converts quantifiers in ghost contexts into mathematical

relations without the need of proving finite enumeration of domains or defining an

algorithm of quantification. This enables the use of more permissive quantifiers. All

the predicates are defined as ghost, even when a predicate doesn’t need to use a more

permissive quantifier, because ghost structures may only be called from ghost contexts.

The notion of provability is available once a terminating judgment was defined.

The predicate for provability says that NDMaxHeight is positive for some proofHeight.

This is shown in Listing 2.3.

Listing 2.3: Provability Predicate
1 ghost predicate NDProof(gamma : set<Formula>, phi : Formula)
2 {
3 exists proofHeight : nat
4 :: proofHeight > 0
5 && NDMaxHeight(gamma, phi, proofHeight)
6 }

2.2 Soundness

Soudness, using naming from the source code, is defined as follows:

NDProof(Γ, φ) =⇒ conseq(Γ, φ),

These are the conditions of the Soundness lemma, as shown in Listing 2.4. The lemma

is proved by first extracting a witness for the proof height (var height :| height

> 0 [...]), and then proceeding to call another lemma, SoundnessMaxHeight, which

inducts on the rules applied that are extracted (there are some rules applied at each

height, increasing, not strictly, in number with height).

Listing 2.4: Soundness lemma

1 lemma Soundness(gamma : set<Formula>, phi : Formula)
2 requires NDProof(gamma, phi)
3 ensures conseq(gamma, phi)
4 {
5 var height :| height > 0 && NDMaxHeight(gamma, phi, height);
6 SoundnessMaxHeight(gamma, phi, height);
7 }

The core lemma of the proof, SoundnessMaxHeight(Γ, ϕ, h), inducts on the

height parameter. At each height some rules of the are applied, so a case for all the

rules is included. For every reference a branch is defined. In a given branch, the the

premises are extracted as facts or as smaller derivability judgements (height -1).

A representative fragment of the implementation is shown in Listing 2.5. The

code shows the overall pattern: detect which rule participated in the derivation at

height height, recursively establish soundness for the premises, and then invoke a

dedicated lemma that determines semantic preservation, that is consequent of the in-

ference rule, such as SoundnessOrElimination.

Listing 2.5: Height-indexed soundness proof (excerpt)
1 lemma SoundnessMaxHeight(gamma : set<Formula>, phi : Formula, height : int)
2 decreases height
3 requires height >= 0
4 requires NDMaxHeight(gamma, phi, height)
5 ensures conseq(gamma, phi)
6 {
7 if(TopIntro(gamma, phi, height)) {
8 forall tau
9 | (forall psi

10 :: psi in gamma ==> value(psi, tau))
11 ensures value(phi, tau)
12 {
13 assert phi == Top;
14 }
15 } else if(Hypothesis(gamma, phi, height)) {
16 forall tau
17 | (forall psi
18 :: psi in gamma ==> value(psi, tau))
19 ensures value(phi, tau)
20 {
21 assert phi in gamma;
22 }
23 } else if(AndIntro(gamma, phi, height)) {
24 var phi1, phi2
25 :| phi == And(phi1, phi2)
26 && NDMaxHeight(gamma, phi1, height -1)
27 && NDMaxHeight(gamma, phi2, height -1);
28 SoundnessMaxHeight(gamma, phi1, height -1);
29 SoundnessMaxHeight(gamma, phi2, height -1);

30 SoundnessAndIntro(gamma, phi1, phi2);
31 } else if(OrElimination(gamma, phi, height)) {
32 var phi1, phi2 :|
33 NDMaxHeight(gamma, Or(phi1, phi2), height -1)
34 && NDMaxHeight(gamma + {phi1}, phi, height -1)
35 && NDMaxHeight(gamma + {phi2}, phi, height -1);
36

37 SoundnessMaxHeight(gamma,Or(phi1, phi2),height-1);
38 SoundnessMaxHeight(gamma+{phi1}, phi, height -1);
39 SoundnessMaxHeight(gamma+{phi2}, phi, height -1);
40

41 SoundnessOrElimination(gamma, phi1, phi2, phi);
42 }
43 // analogous branches for the remaining rules
44 }

There are two base cases: one is for the Top constant top and one for premises

check. In the TopIntro branch, the sought formula must be Top, whose semantic value

is always true. In the Hypothesis branch, the formula is already a member of the con-

text, so any valuation satisfying all formulas in Γ must satisfy it as well. All remaining

cases use the induction hypothesis on strictly smaller heights.

For an example, conjunction introduction is captured by the lemma in Listing 2.6.

It states that if both conjuncts are logical consequences of Γ, then their conjunction is

also a logical consequence of Γ. The body of the proof simply unfolds the definition

of conseq: for an arbitrary truth assignment satisfying the context, the induction hy-

potheses yield truth of both conjuncts, and therefore truth of their conjunction.

Listing 2.6: Semantic preservation of conjunction introduction
1 lemma SoundnessAndIntro(gamma : set<Formula>, phi : Formula, phi’ : Formula)
2 requires conseq(gamma, phi)
3 requires conseq(gamma, phi’)
4 ensures conseq(gamma, And(phi, phi’))
5 {
6 forall tau | (forall psi :: psi in gamma ==> value(psi, tau))
7 ensures value(And(phi, phi’), tau)
8 {
9 assert value(phi, tau);

10 assert value(phi’, tau);
11 assert value(And(phi, phi’), tau);
12 }
13 }

A slightly more interesting case is disjunction elimination, shown in Listing 2.7.

Here the semantic proof follows the intended logical reading of case analysis. If Γ |=

ϕ ∨ ϕ′ and both Γ ∪ {ϕ} |= ψ and Γ ∪ {ϕ′} |= ψ, then under any valuation satisfying Γ,

the truth-hood of ϕ ∨ ϕ′ guarantees that one of the two cases holds. In either case, the

corresponding extended-context assumption yields ψ.

Listing 2.7: Semantic preservation of disjunction elimination
1 lemma SoundnessOrElimination(gamma : set<Formula>, phi : Formula, phi’ : Formula,

psi : Formula)
2 requires conseq(gamma, Or(phi, phi’))
3 requires conseq(gamma + {phi}, psi)
4 requires conseq(gamma + {phi’}, psi)
5 ensures conseq(gamma, psi)
6 {
7 forall tau | (forall chi :: chi in gamma ==> value(chi, tau))
8 ensures value(psi, tau)
9 {

10 assert value(Or(phi, phi’), tau);
11 if value(phi, tau) {
12 assert (forall chi :: chi in gamma + {phi} ==> value(chi, tau));
13 assert value(psi, tau);
14 } else {
15 assert (forall chi :: chi in gamma + {phi’} ==> value(chi, tau));
16 assert value(psi, tau);
17 }
18 }
19 }

Not every rule requires its own auxiliary lemma. In the branches for OrIntro1

and OrIntro2, the proof uses the induction hypothesis together with a direct assertion

that the corresponding disjunction is a consequence of the context. The Dafny system

can discharge these cases without the need to use auxiliary lemmas.

Overall, the soundness proof follows faithfully the standard mathematical argu-

ment that every inference rule of the proof system preserves truth.

2.3 Completeness

To prove completeness, we closely follow the approach proposed by Bruni [1].

We prove the statement

conseq(Γ, β) =⇒ NDProof(Γ, β)

as a Dafny lemma, shown in Listing 2.8.

The proof consists of three steps: (1) move the initial assumptions into the for-

mula as antecedents of nested implications and showing they’re consequent of empty

context (i.e. lemma SemanticDeduction, discussed in Subsection 2.3.1), (2) show that

any tautology is provable (lemma ProvableTautologies, Subsection 2.3.2), (3) move

the antecedents of the implication back as assumptions of a sequent (lemma ChainImplicationToAssumptions,

Subsection 2.3.3).

Listing 2.8: Top-level completeness theorem
1 lemma Completeness(gamma : set<Formula>, beta : Formula)
2 requires conseq(gamma, beta)
3 ensures NDProof(gamma, beta)
4 {
5 var sigma := setToSeq(gamma);
6 SetToSeqProperties(gamma);
7

8 SemanticDeduction(sigma, beta);
9 assert conseq({}, chainImplication(sigma, beta));

10

11 var chainFormula := chainImplication(sigma, beta);
12 assert forall tau :: value(chainFormula, tau);
13

14 if |getFormulaAtoms(chainFormula)| > 0 {
15 ProvableTautologies(chainFormula);
16 } else {
17 var tau := (x: string) => true;
18 ProvableTautologiesSublemma(chainFormula, tau);
19 assert tauToFormulae({}, tau) == {};
20 assert NDProof({}, chainFormula);
21 }
22

23 ChainImplicationToAssumptions(sigma, beta);
24 assert NDProof(seqToSet(sigma), beta);
25 }

The top-level completeness lemma orchestrates the main components of the com-

pleteness argument. Starting from the semantic assumption that beta follows from

gamma, the proof first fixes an ordering of the assumptions by converting the set gamma

into a sequence sigma. It then internalizes these assumptions as a chain of implica-

tions, proves that this chained formula is a tautology, obtains a derivation of it from

the empty context, and finally converts the antecedents of the implication chain back

into explicit assumptions.

2.3.1 Logical Consequences to Tautologies

The first stage of the proof is a semantic deduction step. Given a finite list of

assumptions

σ = [α1, . . . , αn],

and a conclusion β, a Dafny function creates the chained implication

α1 → (α2 → · · · (αn → β) · · ·).

This construction is implemented by the recursive function chainImplication. The

key lemma, SemanticDeduction, states that if β is a logical consequence of the set of

assumptions represented by σ, then the corresponding chained implication is a tautol-

ogy:

conseq(seqToSet(σ), β) =⇒

conseq(∅, chainImplication(σ, β)).

Its implementation is shown in Listing 2.9 and the proof is by contradiction. A

truth assignment negating the chained implication is assumed to exist because of the

tuatology negation and extracted. In order for an implication to be false, its’ antecedent

must be true and its’ consequent false. Lemma ChainImplicationFalseImpliesAllTrue

recurses on each antecedent of the chain implication with the premise that the sub-

structured chain implication is false and ensures that all assumptions are true and beta

false.

Listing 2.9: Semantic deduction lemma
1 lemma SemanticDeduction(sigma : seq<Formula>, beta: Formula)
2 requires conseq(seqToSet(sigma), beta)
3 ensures conseq({}, chainImplication(sigma, beta))
4 {
5 if !conseq({}, chainImplication(sigma, beta)) {
6 var tau : TruthAssignment
7 :| !value(
8 chainImplication(sigma, beta),
9 tau);

10

11 ChainImplicationFalseImpliesAllTrue(sigma, beta, tau);
12 SequenceElementsAreInSet(sigma, tau);
13

14 assert forall phi
15 :: phi in seqToSet(sigma)
16 ==> value(phi, tau);
17 assert !value(beta, tau);
18 assert false;
19 }
20 }

One technical difficulty arising from the use of Dafny in this step is that sets must

be converted to sequences (and back), which introduces some book-keeping in the

proof, which is not conceptual in nature. This is due to the fact that the assumptions in

a sequent are not ordered (they are represented as a set), but in a chained implication

an order must be fixed. Some verification conditions, like the fact that every element

in the initial set occurs (once) in the resulting sequence, must be proven explicitly in

Dafny, where they would be considered immediate or trivial in paper proofs.

Listing 2.10: seqToSet specialized book-keeping for SemanticDeduction
1 lemma SequenceElementsAreInSet(sigma: seq<Formula>, tau: TruthAssignment)
2 requires forall i :: 0 <= i < |sigma| ==> value(sigma[i], tau)
3 ensures forall phi :: phi in seqToSet(sigma) ==> value(phi, tau)
4 decreases |sigma|
5 {
6 if |sigma| == 0 {
7 assert seqToSet(sigma) == {};
8 } else {
9 // seqToSet(sigma) == seqToSet(sigma[1..]) + {sigma[0]}

10 SequenceElementsAreInSet(sigma[1..], tau);
11

12 forall phi | phi in seqToSet(sigma)
13 ensures value(phi, tau)
14 {
15 if phi == sigma[0] {
16 assert value(phi, tau);
17 } else {
18 assert phi in seqToSet(sigma[1..]);
19 assert value(phi, tau);
20 }
21 }
22 }
23 }

2.3.2 Provability of Tautologies

The next step in the proof consists of showing that every tautology is provable.

This is shown in Lemma ProvableTautologies, presented in Listing 2.11.

Listing 2.11: Provability of valid formulas
1 lemma ProvableTautologies(phi: Formula)
2 requires valid(phi)
3 ensures NDProof({}, phi)
4 {
5 var atomsLeft := setToSeq(getFormulaAtoms(phi));
6 PTRecursive(phi, [], atomsLeft);
7 }

The proof is constructive and proceeds by systematically enumerating assign-

ments to the atoms of the formula. The auxiliary function getFormulaAtoms extracts the

propositional variables occurring in a formula, and the recursive lemma PTRecursive

incrementally builds a finite "atomic line" consisting of literals, that is, formulas of the

form p or ¬p. Intuitively, this atomic line represents a partial valuation. At each recur-

sive step, one remaining variable is chosen and the procedure branches into two cases:

one branch adds the positive literal, and the other adds its negation. At the leaves of

recursion all truth assignments that impact the formula are considered.

When no atoms remain, the atomic line completely determines a valuation. The

function buildTauFromAtomicLine reconstructs the corresponding truth assignment,

and the lemma ProvableTautologiesSublemma proves an expected local property: for

any valuation τ , either the formula itself or its negation is derivable from the set of

literals that encode τ . Since PTRecursive is invoked only under the assumption that ϕ

is valid, the base case always obtains a derivation of ϕ, never merely of its negation.

A representative fragment of the recursive construction is shown in Listing 2.12.

The code illustrates the two essential ideas: recursively derive ϕ under both exten-

sions of the current partial assignment, and then combine the two branches using a

machine-checked derivation of the Law of Excluded Middle together with disjunction

elimination. 200+ of lines for book-keeping assertions and calls on helper lemmas were

not included in the listing.

Listing 2.12: Recursive tautology construction (excerpt)
1 lemma PTRecursive(phi : Formula, atomicLine : seq<Formula>, atomsLeft : seq<

Formula>)
2 decreases |atomsLeft|
3 requires valid(phi)
4 ensures NDProof(seqToSet(atomicLine), phi)
5 {
6 if(atomsLeft == []) {
7 var builtTau := buildTauFromAtomicLine(atomicLine);
8 ProvableTautologiesSublemma(phi, builtTau);
9 assert NDProof(seqToSet(atomicLine), phi);

10 } else {
11 var newPositiveAtomicLine := atomicLine + [atomsLeft[0]];
12 var newNegativeAtomicLine := atomicLine + [Not(atomsLeft[0])];
13

14 PTRecursive(phi, newPositiveAtomicLine, atomsLeft[1..]);
15 PTRecursive(phi, newNegativeAtomicLine, atomsLeft[1..]);
16

17 LawExcludedMiddleFormula(atomsLeft[0]);
18 assert NDMaxHeight(seqToSet(atomicLine),
19 Or(atomsLeft[0], Not(atomsLeft[0])),
20 proofHeightMax);
21 assert NDMaxHeight(seqToSet(atomicLine) + {atomsLeft[0]}, phi, proofHeightMax)

;
22 assert NDMaxHeight(seqToSet(atomicLine) + {Not(atomsLeft[0])}, phi,

proofHeightMax);

23

24 assert OrElimination(seqToSet(atomicLine), phi, proofHeightMax + 1);
25 assert NDProof(seqToSet(atomicLine), phi);
26 }
27 }

The recursive lemma PTRecursive is responsible for the constructive proof that

every valid formula is derivable. It works by enumerating all possible truth assign-

ments to the atoms of the formula. The parameter atomicLine records the literals al-

ready chosen for the current partial assignment, while atomsLeft contains the variables

that remain to be assigned. For each remaining atom, the proof recursively handles

both cases—assuming the atom and assuming its negation—and then combines the

two derivations using excluded middle and disjunction elimination. In this way, the

recursion gradually removes all case assumptions and produces a proof from the orig-

inal context.

The branch-combination step is the proof-theoretic heart of the construction. Sup-

pose the current partial context is ∆, and p is the next atom to be assigned. By the re-

cursive calls, one has derivations of ϕ from ∆∪{p} and from ∆∪{¬p}. Independently,

the lemma LawExcludedMiddleFormula provides a derivation of p ∨ ¬p from the empty

context, which is then weakened to ∆. Disjunction elimination then yields a deriva-

tion of ϕ from ∆ itself. Repeating this process until all variables have been discharged

produces a derivation from the empty context, proving the tautology.

Several technical lemmas are needed. In addition to the lemmas relating sets and

sequences, we also require lemmas like ProofHeightExtensibility, which synchro-

nizes derivations to a common height so that inference rules with multiple premises

can be applied. The lemma Weakening is used repeatedly to transport derivations into

larger contexts. Finally, predicates such as noContradictions guarantee that the atomic

line never contains both a variable and its negation, ensuring that the valuation recon-

structed from it is well defined.

2.3.3 From Chained Implications Back to Assumptions

After obtaining a derivation of the chained implication in the empty context, the

final step is to recover a derivation of β from the original assumptions:

NDProof(∅, chainImplication(σ, β))

=⇒ NDProof(seqToSet(σ), β).

This is in Lemma ChainImplicationToAssumptions, shown in Listing 2.14.

Listing 2.13: From chained implications to assumptions
1 lemma ChainImplicationToAssumptions(
2 sigma: seq<Formula>, beta: Formula)
3 requires NDProof({}, chainImplication(sigma, beta))
4 ensures NDProof(seqToSet(sigma), beta)
5 {
6 var proofHeight
7 :| proofHeight > 0
8 && NDMaxHeight({}, chainImplication(sigma, beta),
9 proofHeight);

10 ChainImplicationToAssumptionsHelper(sigma, beta,
11 proofHeight);
12 }

The proof proceeds by induction on the length of the sequence σ. If the sequence

is empty,chainImplication(σ,β) is just β, and the result is immediate. Otherwise, if

chainImplication(σ, β) = α→ chainImplication(σ′, β),

the development weakens the proof into the context {α}, derives α by hypothesis, then

applies implication elimination to obtain proof of the shorter chain chainImplication(σ′,β)

under assumption α. The induction hypothesis is then applied recursively. This closely

mirrors the standard paper argument of repeatedly “peeling off” implications and

turning them into open assumptions.

Listing 2.14: ChainImplicationToAssumptions Helper Lemmas
1 lemma ChainImplicationToAssumptionsHelper(sigma: seq<Formula>, beta: Formula,

proofHeight: nat)
2 decreases |sigma|
3 requires NDMaxHeight({}, chainImplication(sigma, beta), proofHeight)
4 ensures NDProof(seqToSet(sigma), beta)
5 {
6 if |sigma| == 0 {
7 assert chainImplication(sigma, beta) == beta;
8 assert NDMaxHeight({}, beta, proofHeight);
9 Weakening({}, seqToSet([]), beta, proofHeight);

10 assert NDProof(seqToSet(sigma), beta);
11 } else {
12 var alpha := sigma[0];
13 var rest := sigma[1..];
14

15 assert chainImplication(sigma, beta) == Implies(alpha, chainImplication(rest,
beta));

16

17 Weakening({}, {alpha}, Implies(alpha, chainImplication(rest, beta)),
proofHeight);

18 assert {} + {alpha} == {alpha};

19 assert NDMaxHeight({alpha}, Implies(alpha, chainImplication(rest, beta)),
proofHeight);

20

21 assert NDMaxHeight({alpha}, alpha, 1) by {
22 assert Hypothesis({alpha}, alpha, 1);
23 }
24

25 var maxHeight := max(proofHeight, 1);
26 ProofHeightExtensibility({alpha}, Implies(alpha, chainImplication(rest, beta))

, proofHeight, maxHeight);
27 ProofHeightExtensibility({alpha}, alpha, 1, maxHeight);
28

29 assert ImpliesElimination({alpha}, chainImplication(rest, beta), maxHeight +
1);

30 assert NDMaxHeight({alpha}, chainImplication(rest, beta), maxHeight + 1);
31

32 ChainImplicationToAssumptionsHelperWithContext({alpha}, rest, beta, maxHeight
+ 1);

33

34 assert sigma == [alpha] + rest;
35 seqToSetProperty([], alpha, alpha);
36 assert seqToSet([alpha]) == {alpha};
37

38 assert seqToSet(sigma) == seqToSet([alpha] + rest);
39 assert seqToSet([alpha] + rest) == seqToSet(rest) + {alpha} by {
40 seqToSetProperty(rest, alpha, alpha);
41 }
42 assert {alpha} + seqToSet(rest) == seqToSet(rest) + {alpha};
43

44 assert NDProof(seqToSet(sigma), beta);
45 }
46 }
47

48 lemma ChainImplicationToAssumptionsHelperWithContext(gamma: set<Formula>, sigma:
seq<Formula>, beta: Formula, proofHeight: nat)

49 decreases |sigma|
50 requires NDMaxHeight(gamma, chainImplication(sigma, beta), proofHeight)
51 ensures NDProof(gamma + seqToSet(sigma), beta)
52 {
53 if |sigma| == 0 {
54 assert chainImplication(sigma, beta) == beta;
55 Weakening(gamma, gamma + seqToSet([]), beta, proofHeight);
56 assert NDProof(gamma + seqToSet(sigma), beta);
57 } else {
58 var alpha := sigma[0];
59 var rest := sigma[1..];
60

61 assert chainImplication(sigma, beta) == Implies(alpha, chainImplication(rest,
beta));

62

63 var gammaAlpha := gamma + {alpha};
64 Weakening(gamma, gammaAlpha, Implies(alpha, chainImplication(rest, beta)),

proofHeight);
65

66 assert NDMaxHeight(gammaAlpha, alpha, 1) by {
67 assert Hypothesis(gammaAlpha, alpha, 1);
68 }
69

70 var maxHeight := max(proofHeight, 1);
71 ProofHeightExtensibility(gammaAlpha, Implies(alpha, chainImplication(rest,

beta)), proofHeight, maxHeight);
72 ProofHeightExtensibility(gammaAlpha, alpha, 1, maxHeight);
73

74 assert ImpliesElimination(gammaAlpha, chainImplication(rest, beta), maxHeight
+ 1);

75 assert NDMaxHeight(gammaAlpha, chainImplication(rest, beta), maxHeight + 1);
76

77 ChainImplicationToAssumptionsHelperWithContext(gammaAlpha, rest, beta,
maxHeight + 1);

78

79 assert gamma + seqToSet(sigma) == gammaAlpha + seqToSet(rest) by {
80 assert sigma == [alpha] + rest;
81 seqToSetProperty([], alpha, alpha);
82 }
83 }
84 }

Chapter 3

Findings in LLM-assisted Program

Verification Development

This chapter summarizes the findings obtained while using generative AI during

the development of the Dafny formalization described in the previous chapter. The

observations come from the same case study: proving soundness and completeness

for natural deduction in propositional logic using a deep embedding of formulae, se-

mantic consequence, and machine-checked proof obligations. The goal was not to run

a controlled benchmark of language models, but to understand where a modern code

assistant can help in an auto-active verification project and where manual proof engi-

neering remains necessary.

In the project, generative AI was used mainly through GitHub Copilot, with

Claude Sonnet 4.5 being the model that proved most useful in practice. Other models

were tried during preliminary experimentation, including o1, o3-mini, Gemini 2.5 Pro,

and GPT-based models. The role of the assistant was exploratory: it was asked to draft

code, refactor definitions, suggest proof decompositions, and fill in parts of Dafny lem-

mas. The final formalization still required human inspection and correction, especially

because small changes in a Dafny specification can invalidate many downstream proof

obligations.

3.1 General Assessment

The overall experience was mixed. LLM assistance did not replace the work of de-

signing the proof, choosing invariants, or understanding why a particular verification

condition failed. However, it did accelerate several implementation phases. The most

useful outputs were not complete machine-checked proofs, but structurally plausible

proof skeletons, helper lemmas, and local transformations that reduced the amount of

routine Dafny code that had to be written manually.

This distinction is important in auto-active verification. A conventional program

can often be tested after a generated edit, whereas a Dafny proof has to satisfy the

verifier exactly. A fragment that is logically close to the intended argument can still fail

because a quantifier trigger is missing, a recursive call does not decrease, a context has

the wrong shape, or a lemma has been strengthened in a way that breaks its callers.

For this reason, the assistant was most valuable when it helped with structure, and

least valuable when it was expected to discover the final low-level verification steps

unaided.

The development team also had limited prior experience with Dafny. In that

setting, LLM assistance was useful as a way to explore idiomatic proof patterns. It

provided examples of how recursive lemmas, explicit case splits, and assertion chains

are commonly expressed in Dafny. Nevertheless, the generated code had to be treated

as a draft rather than as a trusted proof artifact.

3.2 Observed Strengths

3.2.1 Proof Skeletons and Case Analysis

One recurring strength was the ability to infer the case structure of inductive or

height-indexed proofs. In the soundness proof, the central lemma SoundnessMaxHeight

proceeds by induction on a height parameter and then performs case analysis on the

disjunctive definition of NDMaxHeight. Since each branch of NDMaxHeight corresponds to

one natural deduction rule, the high-level structure of the proof is regular: identify the

active rule, extract its premises, invoke the induction hypothesis on smaller derivations

when necessary, and prove semantic preservation for that rule.

The assistant was able to generate skeletons of this kind. A simplified fragment

is shown in Listing 3.1. The important point is not that the listing proves soundness

by itself, but that it captures the branching pattern that the human developer then

completed.

Listing 3.1: AI-assisted skeleton for a height-indexed soundness proof
1 lemma SoundnessMaxHeight(gamma : set<Formula>, phi : Formula, height : int)
2 decreases height
3 requires height >= 0
4 requires NDMaxHeight(gamma, phi, height)
5 ensures conseq(gamma, phi)
6 {
7 if TopIntro(gamma, phi, height) {
8 forall tau | (forall psi :: psi in gamma ==> value(psi, tau))
9 ensures value(phi, tau)

10 {
11 assert phi == Top;
12 }
13 } else if Hypothesis(gamma, phi, height) {
14 forall tau | (forall psi :: psi in gamma ==> value(psi, tau))
15 ensures value(phi, tau)
16 {
17 assert phi in gamma;
18 }
19 }
20 // additional branches corresponding to other inference rules
21 }

This kind of generated structure is helpful because the proof contains many sim-

ilar branches. The branch for conjunction introduction, for example, has to recover

derivations of both conjuncts and apply the induction hypothesis twice. The branch

for disjunction elimination has to reason under two extended contexts. Even when the

generated code did not finish the proof, it often placed the proof attempt in the right

local configuration.

The assistant also occasionally identified cases in which the verifier needed an

explicit contradiction branch, for instance by using assert false; after incompatible

assumptions had been established. This is a small but practical point: in paper proofs

such contradictions are often implicit, while Dafny may require the impossible branch

to be closed explicitly.

3.2.2 Auxiliary Data-Structure Lemmas

A second area where LLM assistance was useful was the construction of routine

helper functions and simple structural lemmas. The completeness proof requires mov-

ing between sets of assumptions and ordered sequences of assumptions. Sets are the

natural representation for contexts, but sequences are convenient when constructing

chained implications. This mismatch introduces technical obligations that are not cen-

tral to the mathematics, such as proving that elements of a sequence occur in the set

obtained from it.

The assistant was generally reliable for this kind of local recursive code. List-

ing 3.2 shows a representative helper function together with the corresponding mem-

bership lemma.

Listing 3.2: Helper function and invariant lemma for sequence-to-set conversion
1 function seqToSet(S: seq<Formula>): set<Formula>
2 decreases |S|
3 ensures forall f :: f in S ==> f in seqToSet(S)
4 {
5 if |S| == 0 then
6 {}
7 else
8 seqToSet(S[1..]) + { S[0] }
9 }

10

11 lemma seqElementInSet(S: seq<Formula>)
12 ensures forall f :: f in S ==> f in seqToSet(S)
13 decreases |S|
14 {
15 if |S| == 0 {
16 } else {
17 seqElementInSet(S[1..]);
18 forall f | f in S
19 ensures f in seqToSet(S)
20 {
21 if f == S[0] {
22 assert f in seqToSet(S);
23 } else {
24 assert f in S[1..];
25 assert f in seqToSet(S[1..]);
26 assert f in seqToSet(S);
27 }
28 }
29 }
30 }

These lemmas are typical examples of where generative AI can save time. They

are standard recursive arguments, have short specifications, and do not require dis-

covering a new proof idea. In contrast to the main soundness and completeness ar-

guments, the proof burden is mostly bookkeeping. For such tasks, the assistant can

provide a useful first draft that is close to the required Dafny style.

3.2.3 Decomposition of Larger Lemmas

The assistant was also useful for decomposing some larger lemmas into smaller

proof obligations. The clearest example was the final step of the completeness proof,

where a proof of a chained implication from the empty context has to be converted back

into a proof from the original assumptions. This is the lemma shown in Listing 2.14.

The generated structure matched the intended mathematical argument: take the

first antecedent of the implication chain, weaken the proof into a context containing

that antecedent, derive the antecedent by hypothesis, apply implication elimination,

and continue recursively on the remaining chain. This was not a complete solution to

all verification details, but it was a useful outline of the proof. In this case, the assistant

helped express the proof strategy in Dafny’s lemma-oriented style.

This observation suggests a useful pattern: LLMs can be effective when a proof

has a clear recursive or syntactic decomposition, even if the individual verification con-

ditions still require manual work. The assistant can help expose the shape of the proof,

while the developer remains responsible for checking that the generated auxiliary lem-

mas are both true and strong enough for the final theorem.

3.3 Observed Limitations

3.3.1 Fragile Context for Dafny Tasks

The effective context window for Dafny development was fragile. After a few

interactions, several models began to produce changes that were either too small to

be useful or not syntactically valid Dafny. In preliminary experimentation, o1, o3-

mini, and Gemini 2.5 Pro often failed early in a session. GPT-based models were more

robust, but they could take a long time to analyze nontrivial proof fragments and still

tended to introduce syntax errors after repeated iterations. Claude Sonnet 4.5, accessed

through GitHub Copilot, gave the best practical results in this case study.

The likely reason is that Dafny proof code combines programming syntax, logical

specifications, SMT-oriented assertions, and mathematical proof structure. A model

has to preserve all of these layers simultaneously. Losing track of any one of them can

produce code that looks plausible but does not verify.

3.3.2 Incomplete Proofs and Auxiliary Lemmas

For complex goals, the assistant often introduced auxiliary lemmas that decom-

posed the original task into smaller tasks. This was useful when the decomposition re-

flected the intended proof. However, the generated sub-lemmas frequently contained

unfinished proof bodies, including unsound assumptions or comments indicating that

an argument was missing. In those situations, the assistant had not solved the proof; it

had only moved the difficult part into a new lemma.

This behavior can still be valuable if the new lemmas expose a sensible proof ar-

chitecture. It becomes problematic when the generated lemmas are as hard to prove

as the original theorem, or when their statements are not exactly what the surround-

ing development needs. The developer then has to spend additional time deciding

whether to keep the decomposition, weaken or strengthen the helper lemmas, or dis-

card the generated attempt entirely.

3.3.3 Specification Drift

Another recurring failure mode was specification drift. When asked to repair a

partially implemented lemma, the assistant sometimes tried to satisfy the verifier by

adding new requires or ensures clauses. Over several iterations, this could produce

a lemma that verified locally but no longer matched its intended role. For example, a

strengthened precondition might not hold at the original call site, or a modified post-

condition might no longer provide the fact needed by the caller.

This is especially dangerous in Dafny because specifications are part of the proof

interface. A generated edit that changes a lemma signature is not merely an implemen-

tation detail; it changes the theorem being used by the rest of the development. Con-

sequently, any AI-generated modification to a specification must be reviewed more

carefully than a change inside a proof body.

3.3.4 Specialized Bookkeeping in Completeness

The least successful area was specialized bookkeeping in the completeness proof.

The mathematical proof enumerates truth assignments to the atoms of a formula, but

the Dafny development has to explicitly manage sequences of remaining variables,

atomic lines, conversions between data structures, and invariants such as absence of

contradictions. This bookkeeping is much more verbose than the corresponding paper

proof.

Generative AI was less useful for these specialized obligations. The assistant

could often write a plausible recursive function or local lemma, but it struggled to

maintain the global invariant structure across the whole completeness proof. This

suggests that LLM assistance is currently stronger on common proof patterns than on

project-specific invariant management.

3.4 Practical Workflow Lessons

The findings from the case study lead to a practical workflow for using generative

AI in Dafny developments. The assistant should be treated as a proof-structuring tool,

not as an autonomous verifier. The most productive prompts are local and explicit:

ask for a branch structure, a helper lemma, a refactoring that preserves a signature,

or a proof outline for a specific recursive function. Broad prompts such as asking the

model to “finish the completeness proof” tend to produce unfinished or speculative

auxiliary lemmas.

It is also important to verify frequently. Each generated edit should be checked

before the conversation moves far from the original context. If the assistant proposes

new preconditions, changes a lemma statement, or introduces an assume, the change

should be treated as suspect until its role in the larger proof is clear. In practice, the

most useful interaction pattern was iterative: generate a small proof fragment, run the

verifier, inspect the failing obligations, and then either repair the fragment manually

or ask the assistant for a narrowly scoped adjustment.

The main observations can be summarized as follows:

• For inductive proof skeletons, the assistant correctly suggested case splits and

recursive proof structure, but the branches still required manual assertions and

verifier-specific details.

• For auxiliary data lemmas, it produced useful recursive definitions and simple

invariants, but the statements still had to be reviewed against later use sites.

• For larger proof decompositions, it helped identify intermediate lemmas and

proof phases, but sometimes replaced one hard goal with several hard subgoals.

• For failed lemma repair, it suggested possible assertions or missing proof steps,

but could drift by strengthening requires clauses or changing interfaces.

• For completeness bookkeeping, it helped with local sequence or set manipula-

tions, but struggled to preserve global invariants.

3.5 Threats to Validity

These findings are necessarily anecdotal. They are based on a single case study,

one proof system and one verification language. The results may also depend on the

specific model versions, editor integration, prompting style, and level of Dafny experi-

ence of the users. A scientifically stronger evaluation would require replication across

multiple formalization tasks, different logics or proof systems, different representa-

tions of proofs, and developers with different levels of expertise.

Even with this limitation, the case study supports a clear conclusion for this the-

sis. Generative AI can accelerate parts of formal verification development, especially

when the task is local, structural, and similar to common recursive proof patterns. It is

much less reliable as a replacement for proof design, invariant discovery, or final verifi-

cation. In the present formalization, the assistant was most useful as a collaborator for

drafting and restructuring proof code, while the correctness of the final development

remained dependent on human-guided proof engineering and Dafny’s verifier.

Chapter 4

Derivation Tree Natural Deduction

This chapter presents a second Dafny formalization of natural deduction, devel-

oped using explicit derivation trees. Unlike the height-bounded judgment from the

previous formalization, this approach represents derivations directly as a data struc-

ture: each constructor of the proof datatype corresponds to one natural deduction rule

and stores the subderivations needed by that rule. The purpose of this development

was twofold. First, I aimed to obtain a formalization in which proof objects are explicit

and can be inspected structurally. Second, I used the development as a case study for

evaluating how well a Codex-based agent can construct Dafny verification code with

as little manual implementation as possible.

The formalization was developed mainly with the assistance of Codex. The ini-

tial drafting phase consisted of defining the derivation-tree datatype and writing the

signatures of the main predicates and helper methods. After the datatype declaration

and the signatures for NDDTProof and NDDTRecursive were provided, Codex was able to

draft a complete natural deduction system. The later stages were more proof-oriented:

for soundness and completeness, I wrote lemma signatures manually, including the rel-

evant preconditions and postconditions, and then asked Codex to complete the proof

body. This made the experiment focused on proof construction rather than on invent-

ing the specification.

The resulting formalization proves soundness and completeness for the same

propositional logic as the height-bounded development. However, the proof engi-

neering differs significantly. In the height-bounded version, derivability is expressed

indirectly through a recursively defined predicate indexed by proof height. In the

derivation-tree version, derivability is expressed by existence of a valid proof object.

This makes some reasoning more natural, especially soundness, because the induction

principle follows the structure of the tree itself. At the same time, it introduces new

bookkeeping obligations: each constructor stores a context and conclusion formula,

and the validity predicate must ensure that these fields agree with the rule represented

by the node.

4.1 Methodology

The proof structure targeted in this development was Bruni’s proof structure, the

same proof structure followed in the original height-bounded formalization described

earlier. This choice made the two developments comparable: the mathematical proof

plan remained stable, while the representation of derivability changed from a height-

indexed predicate to an explicit derivation-tree datatype.

The development process was split into two stages. In the first stage, the core

data structures were drafted. This included the formula datatype inherited from the

previous formalization, the representation of derivation trees, and the constructors cor-

responding to the natural deduction rules. The goal of this stage was to obtain a coher-

ent representation of proof objects before attempting the main verification lemmas.

In the second stage, Codex was used for proof development. The interaction pat-

tern was deliberately constrained. For each lemma, the lemma signature was written

manually: the name, parameters, theory-level preconditions, and postconditions were

specified. Some basic functions and definitions from the original development were

also provided manually when they were needed to state or guide the proof. Codex

was then prompted to complete the lemma body.

To make the task realistic while still giving the agent access to relevant context,

the original height-bounded formalization was provided as an accessible folder during

Codex sessions. The agent could consult that development for analogous definitions,

proof decomposition patterns, and helper lemmas. However, the target code still had

to be adapted to the derivation-tree representation, since rules were no longer repre-

sented by a height-bounded predicate but by constructors of an explicit proof datatype.

This distinction turned out to be important. Codex was often effective when the

new proof had the same high-level structure as an existing one, but it sometimes copied

the old representation too literally. During the completeness proof, for example, after

the signature of the second lemma and its sublemma were copied over, Codex initially

tried to build a proof tree by calling NDMaxHeight from the previous natural deduction

module. This was incorrect for the derivation-tree formalization, so that attempt was

undone and the prompt was changed to explicitly forbid use of the old basic natural

deduction module. This illustrates a recurring pattern in the development: Codex was

useful when the specification and representation boundaries were made explicit, but it

could drift toward nearby code if the prompt allowed ambiguity.

4.2 Data Structures

The propositional object language is the same as in the previous development.

Formulae are represented as an algebraic datatype with constructors for variables,

negation, conjunction, disjunction, implication, bottom, and top.

Listing 4.1: Formula datatype
1 datatype Formula = Var(name: string)
2 | Not(f: Formula)
3 | And(f1: Formula, f2: Formula)
4 | Or(f1: Formula, f2: Formula)
5 | Implies(f1: Formula, f2: Formula)
6 | Bottom
7 | Top

Truth assignments are total functions from variable names to Boolean values, and

semantic evaluation is defined recursively over formula structure.

Listing 4.2: Truth assignments and semantic evaluation
1 type TruthAssignment = string -> bool
2

3 function value(f: Formula, tau: TruthAssignment) : bool
4 decreases f
5 {
6 match f
7 case Var(x) => tau(x)
8 case Not(p) => !value(p, tau)
9 case And(p, q) => value(p, tau) && value(q, tau)

10 case Or(p, q) => value(p, tau) || value(q, tau)
11 case Implies(p, q) => !value(p, tau) || value(q, tau)
12 case Bottom => false
13 case Top => true
14 }

The new part of this chapter is the proof datatype. A derivation is represented

by the datatype DerivationTree. Each constructor corresponds to one rule of natu-

ral deduction and stores the context, the conclusion formula, and the required child

derivations.

Listing 4.3: Derivation tree datatype
1 datatype DerivationTree =
2 | HypothesisNode(gamma : set<Formula>, phi’ : Formula)
3 | AndIntroNode(gamma : set<Formula>, phi’ : Formula,
4 left : DerivationTree, right : DerivationTree)
5 | AndElimination1Node(gamma : set<Formula>, phi’ : Formula,
6 psi : Formula, child : DerivationTree)
7 | AndElimination2Node(gamma : set<Formula>, phi’ : Formula,
8 psi : Formula, child : DerivationTree)
9 | ImpliesEliminationNode(gamma : set<Formula>, phi’ : Formula,

10 left : DerivationTree, right : DerivationTree)
11 | ImpliesIntroNode(gamma : set<Formula>, phi’ : Formula,
12 phi1 : Formula, phi2: Formula,
13 child : DerivationTree)
14 | OrIntro1Node(gamma : set<Formula>, phi’ : Formula,
15 phi1 : Formula, phi2 : Formula,
16 child : DerivationTree)
17 | OrIntro2Node(gamma : set<Formula>, phi’ : Formula,
18 phi1 : Formula, phi2 : Formula,
19 child : DerivationTree)
20 | OrEliminationNode(gamma : set<Formula>, phi’ : Formula,
21 phi1 : Formula, phi2 : Formula,
22 orProof : DerivationTree,
23 leftProof : DerivationTree,
24 rightProof : DerivationTree)
25 | NotIntroNode(gamma : set<Formula>, phi’ : Formula,
26 phi1 : Formula, child : DerivationTree)
27 | NotEliminationNode(gamma : set<Formula>, phi’ : Formula,
28 psi : Formula,
29 left : DerivationTree, right : DerivationTree)
30 | BottomEliminationNode(gamma : set<Formula>, phi’ : Formula,
31 child : DerivationTree)
32 | NotNotEliminationNode(gamma : set<Formula>, phi’ : Formula,
33 child : DerivationTree)
34 | TopIntroNode(gamma : set<Formula>, phi’ : Formula)

The fields gamma and phi’ represent the root judgment of the node: the node is

intended to prove phi’ from the assumptions in gamma. This information is extracted

by the functions RootGamma and RootFormula.

Listing 4.4: Root judgment accessors
1 function RootGamma(dt : DerivationTree) : set<Formula>
2 decreases dt
3 {
4 match dt
5 case HypothesisNode(gamma, _) => gamma
6 case AndIntroNode(gamma, _, _, _) => gamma
7 ...
8 }
9

35



10 function RootFormula(dt : DerivationTree) : Formula
11 decreases dt
12 {
13 match dt
14 case HypothesisNode(_, phi’) => phi’
15 case AndIntroNode(_, phi’, _, _) => phi’
16 ...
17 }

The height of a derivation tree is defined structurally. Leaf nodes have height

one, unary nodes add one to the height of their child, and binary or ternary rules add

one to the maximum height of their premises.

Listing 4.5: Derivation tree height
1 function DTHeight(dt : DerivationTree) : nat
2 decreases dt
3 {
4 match dt
5 case HypothesisNode(_, _) => 1
6 case AndIntroNode(_, _, left, right) =>
7 1 + max(DTHeight(left), DTHeight(right))
8 case OrEliminationNode(_, _, _, _, orProof, leftProof, rightProof) =>
9 1 + max(DTHeight(orProof),

10 max(DTHeight(leftProof), DTHeight(rightProof)))
11 ...
12 }

Although the formalization is no longer defined by a height-bounded predicate,

the height function remains useful. It allows the development to recover a bounded

notion of derivability when needed, and it also gives a concrete measure of proof size.

The main proof relation, however, is based on validity of explicit proof objects.

4.3 Valid Derivation Trees

The central predicate of the formalization is NDDTValid. It recursively checks

whether a derivation tree is a valid natural deduction proof. Each case corresponds

to one inference rule.

Listing 4.6: Validity predicate excerpt
1 ghost predicate NDDTValid(dt : DerivationTree)
2 decreases dt
3 {
4 match dt
5 case HypothesisNode(gamma, phi’) =>
6 phi’ in gamma
7

8 case AndIntroNode(gamma, phi’, left, right) =>
9 NDDTValid(left) &&

10 NDDTValid(right) &&
11 RootGamma(left) == gamma &&
12 RootGamma(right) == gamma &&
13 phi’ == And(RootFormula(left), RootFormula(right))
14

15 case AndElimination1Node(gamma, phi’, psi, child) =>
16 NDDTValid(child) &&
17 RootGamma(child) == gamma &&
18 RootFormula(child) == And(phi’, psi)
19

20 ...
21 }

The hypothesis rule is the simplest case: a node is valid if its formula is contained

in its context. Conjunction introduction requires two valid subtrees with the same

context, and the root formula of the parent must be the conjunction of the two child

conclusions. Conjunction elimination checks that the child proves a conjunction and

that the parent extracts the appropriate component.

The implication rules show the difference between storing proof structure and

merely defining a derivability predicate. Implication elimination is represented by a

node containing a proof of an implication and a proof of its antecedent.

Listing 4.7: Implication elimination validity case
1 case ImpliesEliminationNode(gamma, phi’, left, right) =>
2 NDDTValid(left) &&
3 NDDTValid(right) &&
4 RootGamma(left) == gamma &&
5 RootGamma(right) == gamma &&
6 RootFormula(left) == Implies(RootFormula(right), phi’)

Here right proves the antecedent, while left proves an implication from that

antecedent to phi’. Therefore, the node proves phi’.

Implication introduction stores both the antecedent and consequent explicitly.

Listing 4.8: Implication introduction validity case
1 case ImpliesIntroNode(gamma, phi’, phi1, phi2, child) =>
2 NDDTValid(child) &&
3 phi’ == Implies(phi1, phi2) &&
4 RootGamma(child) == gamma + {phi1} &&
5 RootFormula(child) == phi2

This corresponds to the standard natural deduction rule: if phi2 can be derived

from gamma extended with the temporary assumption phi1, then Implies(phi1, phi2)

can be derived from gamma.

One design decision was especially important. In an earlier version, several rule

cases used existentially quantified witnesses inside NDDTValid, following the style of

the height-bounded predicate. Following a suggestion from the supervisor, these wit-

nesses were moved into the constructor parameters. For example, the disjunction elim-

ination node stores phi1 and phi2 directly, and implication introduction stores phi1

and phi2 directly. This made the validity predicate more concrete and reduced the

need for Dafny to guess witnesses during proof checking.

This change also made the proof objects more informative. A valid disjunction-

elimination node does not merely state that some disjunction was eliminated; it records

the two disjuncts and the three subderivations involved in the case split.

Listing 4.9: Disjunction elimination validity case
1 case OrEliminationNode(gamma, phi’, phi1, phi2,
2 orProof, leftProof, rightProof) =>
3 NDDTValid(orProof) &&
4 NDDTValid(leftProof) &&
5 NDDTValid(rightProof) &&
6 RootGamma(orProof) == gamma &&
7 RootFormula(orProof) == Or(phi1, phi2) &&
8 RootGamma(leftProof) == gamma + {phi1} &&
9 RootGamma(rightProof) == gamma + {phi2} &&

10 RootFormula(leftProof) == phi’ &&
11 RootFormula(rightProof) == phi’

The proof predicate is then defined by existential quantification over valid deriva-

tion trees.

Listing 4.10: Derivation-tree proof predicate
1 ghost predicate NDDTProof(gamma : set<Formula>, phi : Formula)
2 {
3 exists dt ::
4 NDDTValid(dt) &&
5 RootGamma(dt) == gamma &&
6 RootFormula(dt) == phi
7 }

Thus NDDTProof(gamma, phi) means that there exists an explicit valid proof tree

whose root judgment is Γ ⊢ φ.

For comparison with the previous formalization, the development also defines a

height-bounded variant.

Listing 4.11: Height-bounded view of derivation trees
1 ghost predicate NDDTMaxHeight(
2 gamma : set<Formula>, phi : Formula, maxHeight : nat)

3 {
4 exists dt ::
5 NDDTValid(dt) &&
6 RootGamma(dt) == gamma &&
7 RootFormula(dt) == phi &&
8 DTHeight(dt) <= maxHeight
9 }

This predicate is not the main definition of derivability. Instead, it is a derived

view of proof trees. The helper method NDDTRecursive extracts a witness tree from

such an existential statement.

Listing 4.12: Extracting a bounded witness tree
1 ghost method NDDTRecursive(
2 gamma : set<Formula>, phi : Formula, maxHeight : nat)
3 returns (dt : DerivationTree)
4 requires maxHeight > 0
5 requires NDDTMaxHeight(gamma, phi, maxHeight)
6 ensures NDDTValid(dt)
7 ensures RootGamma(dt) == gamma
8 ensures RootFormula(dt) == phi
9 ensures DTHeight(dt) <= maxHeight

10 {
11 dt :| NDDTValid(dt) &&
12 RootGamma(dt) == gamma &&
13 RootFormula(dt) == phi &&
14 DTHeight(dt) <= maxHeight;
15 }

After the datatype and these signatures were written, Codex was able to generate

the complete natural deduction system. This was one of the most successful parts of the

experiment. The structure of the datatype closely mirrors the mathematical inference

rules, so the model could infer the pattern for most constructors and validity cases

once a few examples were available.

4.4 Soundness

The soundness theorem states that every derivable formula is a semantic conse-

quence of its assumptions.

Listing 4.13: Top-level derivation-tree soundness theorem
1 lemma NDDTSoundness(gamma : set<Formula>, phi : Formula)
2 requires NDDTProof(gamma, phi)
3 ensures conseq(gamma, phi)
4 {
5 var dt : DerivationTree :|

6 NDDTValid(dt) &&
7 RootGamma(dt) == gamma &&
8 RootFormula(dt) == phi;
9 NDDTSoundnessTree(dt);

10 }

The proof first extracts the proof tree witnessing NDDTProof(gamma, phi). Then it

calls the stronger induction lemma NDDTSoundnessTree, which proves soundness for a

concrete derivation tree.

Listing 4.14: Structural soundness lemma
1 lemma NDDTSoundnessTree(dt : DerivationTree)
2 requires NDDTValid(dt)
3 ensures conseq(RootGamma(dt), RootFormula(dt))
4 decreases dt
5 {
6 match dt
7 case HypothesisNode(gamma, phi’) => ...
8 case AndIntroNode(gamma, phi’, left, right) => ...
9 case ImpliesIntroNode(gamma, phi’, phi1, phi2, child) => ...

10 ...
11 }

This is a natural proof by structural induction on the derivation tree. In the hy-

pothesis case, the conclusion is already in the context. Therefore, any truth assignment

satisfying all assumptions must satisfy the conclusion. In the top introduction case,

the conclusion is Top, which is always true. For the remaining cases, the proof re-

cursively applies NDDTSoundnessTree to the child derivations and then proves that the

corresponding semantic rule is valid.

For conjunction introduction, for example, the induction hypotheses show that

both child conclusions are true under any valuation satisfying the context. Since the

parent formula is the conjunction of the two child formulae, the parent formula is true

as well.

Listing 4.15: Soundness of conjunction introduction
1 case AndIntroNode(gamma, phi’, left, right) => {
2 NDDTSoundnessTree(left);
3 NDDTSoundnessTree(right);
4 forall tau | (forall psi :: psi in gamma ==> value(psi, tau))
5 ensures value(phi’, tau)
6 {
7 assert RootGamma(left) == gamma;
8 assert RootGamma(right) == gamma;
9 assert value(RootFormula(left), tau);

10 assert value(RootFormula(right), tau);
11 assert phi’ == And(RootFormula(left), RootFormula(right));

12 }
13 }

The implication introduction case follows the standard semantic argument. To

prove that phi1 -> phi2 is true under a valuation satisfying gamma, Dafny splits on

whether phi1 is true. If it is false, the implication is true immediately. If it is true, then

the valuation also satisfies gamma + {phi1}, and the induction hypothesis for the child

gives truth of phi2.

Listing 4.16: Soundness of implication introduction
1 case ImpliesIntroNode(gamma, phi’, phi1, phi2, child) => {
2 NDDTSoundnessTree(child);
3 assert phi’ == Implies(phi1, phi2)
4 && RootGamma(child) == gamma + {phi1}
5 && RootFormula(child) == phi2;
6 forall tau | (forall psi :: psi in gamma ==> value(psi, tau))
7 ensures value(phi’, tau)
8 {
9 if value(phi1, tau) {

10 assert forall chi :: chi in gamma + {phi1} ==> value(chi, tau);
11 assert value(phi2, tau);
12 }
13 assert value(Implies(phi1, phi2), tau);
14 }
15 }

The most notable difficulty occurred in the double-negation elimination case.

After the signature of NDDTSoundness and its preconditions and postconditions were

written, Codex drafted the main proof and introduced a helper lemma for structural

induction. Most cases followed the expected pattern, but the double-negation elim-

ination node contained an error. The node validity condition gives a child proof of

Not(Not(phi’)); the proof must then establish phi’. Dafny did not automatically

bridge this semantic equivalence in the way the generated proof expected.

The fix was to add the following manually written lemma.

Listing 4.17: Semantic bridge for double negation
1 lemma NotNotFEliminationEquivalent(
2 gamma : set<Formula>, phi : Formula)
3 requires conseq(gamma, Not(Not(phi)))
4 ensures conseq(gamma, phi)
5 {
6 forall tau | (forall psi :: psi in gamma ==> value(psi, tau))
7 ensures value(phi, tau) <==> value(Not(Not(phi)), tau)
8 ensures !value(Not(Not(phi)), tau) <==> !value(phi, tau)
9 {

10 assert value(Not(Not(phi)), tau);

11 }
12 }

The double-negation elimination case could then be completed by calling this

lemma after applying the induction hypothesis to the child derivation.

Listing 4.18: Soundness of double-negation elimination
1 case NotNotEliminationNode(gamma, phi’, child) => {
2 NDDTSoundnessTree(child);
3 forall tau | (forall psi :: psi in gamma ==> value(psi, tau))
4 ensures value(phi’, tau)
5 {
6 assert RootGamma(child) == gamma;
7 assert RootFormula(child) == Not(Not(phi’));
8 assert value(Not(Not(phi’)), tau);
9 NotNotFEliminationEquivalent(gamma, phi’);

10 assert value(phi’, tau);
11 }
12 }

This episode is representative of the development experience. Codex produced a

nearly complete structural proof, but a small semantic bridge still had to be identified

and inserted manually. After the change suggested by the supervisor, namely moving

witnesses into constructor parameters, the soundness proof was also updated to fit the

new datatype structure. The resulting proof is more direct than the height-bounded

version because the recursive calls follow the actual subtrees instead of an external

height argument.

4.5 Completeness

The completeness theorem states the converse of soundness: if a formula is a

semantic consequence of a context, then there exists a derivation tree proving it from

that context.

Listing 4.19: Derivation-tree completeness theorem
1 lemma NDDTCompleteness(gamma : set<Formula>, phi : Formula)
2 requires conseq(gamma, phi)
3 ensures NDDTProof(gamma, phi)

The proof follows the same high-level structure as the previous formalization. It

consists of three stages.

First, the assumptions are internalized into a chain of implications. A set of as-

sumptions gamma is converted into a sequence sigma, and the formula

α1 → α2 → · · · → φ

is constructed using chainImplication.

Listing 4.20: Chained implication formula
1 function chainImplication(S: seq<Formula>, beta: Formula): Formula
2 decreases |S|
3 {
4 if |S| == 0 then
5 beta
6 else
7 Implies(S[0], chainImplication(S[1..], beta))
8 }

The first completeness lemma shows that if beta is a consequence of the set of

formulae in sigma, then the chained implication is valid in the empty context.

Listing 4.21: Semantic deduction for chained implications
1 lemma NDDTSemanticDeduction(sigma : seq<Formula>, beta: Formula)
2 requires conseq(seqToSet(sigma), beta)
3 ensures conseq({}, chainImplication(sigma, beta))

This lemma is essentially the same as in the height-bounded formalization and

was copied over with only minor adaptation. Its proof proceeds by contradiction: if

the chained implication is false under some truth assignment, then all antecedents in

the sequence must be true and beta must be false. This contradicts the semantic con-

sequence assumption.

Second, the development proves that valid formulae are derivable. This is the

most involved part of completeness. The key sublemma is valuation-relative.

Listing 4.22: Valuation-relative tautology sublemma
1 lemma ProvableTautologiesSublemma(phi: Formula, tau: TruthAssignment)
2 ensures value(phi, tau) ==>
3 NDDTProof(tauToFormulae(getFormulaAtoms(phi), tau), phi)
4 ensures !value(phi, tau) ==>
5 NDDTProof(tauToFormulae(getFormulaAtoms(phi), tau), Not(phi))

The context tauToFormulae(getFormulaAtoms(phi), tau) contains one literal for

each atom of phi, chosen according to the truth assignment tau. If an atom is true

under tau, the context contains the atom itself; if it is false, the context contains its

negation. The sublemma states that, from the literals describing a valuation, one can

derive either the formula or its negation depending on its truth value.

This sublemma is proved by structural case analysis on phi. The base case for

variables is immediate: the valuation context already contains either the variable or its

negation. The cases for implication, conjunction, and disjunction build explicit deriva-

tion trees using the natural deduction constructors.

The negation case was one of the main manual proof-engineering points. When

phi = Not(phi’) and value(phi, tau) is false, then phi’ is true. The desired conclusion

is a proof of Not(phi), which is Not(Not(phi’)). The proof first derives phi’ from the

valuation context, then temporarily assumes Not(phi’), derives a contradiction, and

discharges the assumption using negation introduction.

Listing 4.23: Negation case of the tautology sublemma
1 lemma ProvableTautologiesSublemmaNotCase(
2 phi: Formula, phi’: Formula, tau: TruthAssignment)
3 requires phi == Not(phi’)
4 ensures value(phi, tau) ==>
5 NDDTProof(tauToFormulae(getFormulaAtoms(phi), tau), phi)
6 ensures !value(phi, tau) ==>
7 NDDTProof(tauToFormulae(getFormulaAtoms(phi), tau), Not(phi))
8 {
9 var formulae := tauToFormulae(getFormulaAtoms(phi), tau);

10 var formulae’ := tauToFormulae(getFormulaAtoms(phi’), tau);
11 assert formulae == formulae’;
12

13 if value(phi, tau) {
14 assert value(phi’, tau) == false;
15 ProvableTautologiesSublemma(phi’, tau);
16 assert NDDTProof(formulae’, Not(phi’));
17 assert phi == Not(phi’);
18 assert NDDTProof(formulae, phi);
19 } else {
20 assert value(phi’, tau) == true;
21 ProvableTautologiesSublemma(phi’, tau);
22

23 assert NDDTProof(formulae, phi’);
24 WeakenProof(formulae, phi’, {Not(phi’)});
25 assert NDDTProof(formulae + {Not(phi’)}, phi’);
26

27 var dtNotPhi’ :=
28 HypothesisNode(formulae + {Not(phi’)}, Not(phi’));
29

30 var dtPhi’ : DerivationTree :|
31 NDDTValid(dtPhi’) &&
32 RootGamma(dtPhi’) == formulae + {Not(phi’)} &&
33 RootFormula(dtPhi’) == phi’;
34

35 var dtBottom :=
36 NotEliminationNode(formulae + {Not(phi’)},
37 Bottom, phi’, dtNotPhi’, dtPhi’);
38

39 var dtNotNotPhi’ :=
40 NotIntroNode(formulae, Not(Not(phi’)),
41 Not(phi’), dtBottom);
42

43 assert Not(phi) == Not(Not(phi’));
44 assert NDDTProof(formulae, Not(phi));
45 }
46 }

This proof was first drafted manually up to the point where the needed derivation

tree structure became clear. Codex then completed the lemma by following the analo-

gous proof pattern. After that, the implication, conjunction, and disjunction cases were

completed successfully by Codex, because the structure of the negation case and the

analogous height-bounded proof gave it enough guidance. The bottom case was also

manually drafted, while Codex completed the top case.

The third part of the tautology proof enumerates all possible truth assignments

over the atoms of the formula. This is implemented by PTRecursive.

Listing 4.24: Recursive valuation enumeration
1 lemma PTRecursive(
2 phi: Formula,
3 atomicLine: seq<Formula>,
4 atomsLeft: seq<Formula>)
5 requires |getFormulaAtoms(phi)| > 0
6 requires valid(phi)
7 ...
8 ensures NDDTProof(seqToSet(atomicLine), phi)
9 decreases |atomsLeft|

The sequence atomicLine represents the literals chosen so far, while atomsLeft

contains the atoms that still need to be assigned. If no atoms are left, a truth assignment

is built from atomicLine, and the valuation-relative sublemma proves phi from the

corresponding literal context.

If atoms remain, the proof branches on the next atom. One recursive call adds the

atom positively, and the other adds its negation.

Listing 4.25: Branching on the next atom
1 var newPositiveLine := atomicLine + [atomsLeft[0]];
2 PTRecursive(phi, newPositiveLine, newAtomsLeft);
3

4 var newNegativeLine := atomicLine + [Not(atomsLeft[0])];
5 PTRecursive(phi, newNegativeLine, newAtomsLeft);

The two branches produce derivations of phi under contexts extended respec-

tively with p and Not(p). To remove this case split, the formalization proves the law

of excluded middle in the derivation-tree system and then applies disjunction elimina-

tion.

Listing 4.26: Combining valuation branches by disjunction elimination
1 NDDTLawExcludedMiddleContext(seqToSet(atomicLine), atomsLeft[0]);
2

3 var lemDt : DerivationTree :=
4 OrEliminationNode(seqToSet(atomicLine),
5 phi,
6 atomsLeft[0],
7 Not(atomsLeft[0]),
8 orDt,
9 posdt,

10 negdt);
11

12 assert NDDTValid(lemDt);
13 assert RootGamma(lemDt) == seqToSet(atomicLine);
14 assert RootFormula(lemDt) == phi;

This is the proof-theoretic heart of completeness. The derivation tree explicitly

records the excluded-middle proof and the two branch derivations, and the validity

predicate checks that the contexts and conclusions match the rule.

Finally, the third major lemma converts a proof of the chained implication back

into a proof from the original assumptions.

Listing 4.27: From chained implications to assumptions
1 lemma ChainImplicationToAssumptions(sigma: seq<Formula>, beta: Formula)
2 requires NDDTProof({}, chainImplication(sigma, beta))
3 ensures NDDTProof(seqToSet(sigma), beta)

The auxiliary lemma works recursively over the sequence of assumptions. At

each step, it weakens the proof of the implication chain into a context containing the

current assumption, derives the current assumption by hypothesis, and applies impli-

cation elimination to obtain the shorter chain.

Listing 4.28: One step of eliminating a chained implication
1 var dtTail :=
2 ImpliesEliminationNode(targetGamma, tail, dtImp, dtPhi);
3

4 assert NDDTValid(dtTail);
5 assert RootGamma(dtTail) == targetGamma;
6 assert RootFormula(dtTail) == tail;
7 assert NDDTProof(targetGamma, tail);
8

9 ChainImplicationToAssumptionsAux(targetGamma, rest, beta);

The top-level completeness theorem combines these three stages.

Listing 4.29: Top-level completeness proof
1 lemma NDDTCompleteness(gamma : set<Formula>, phi : Formula)
2 requires conseq(gamma, phi)
3 ensures NDDTProof(gamma, phi)
4 {
5 var sigma := setToSeq(gamma);
6 setSeqSetEquality(gamma);
7 assert seqToSet(sigma) == gamma;
8

9 NDDTSemanticDeduction(sigma, phi);
10 assert conseq({}, chainImplication(sigma, phi));
11

12 var chainFormula := chainImplication(sigma, phi);
13 assert forall tau :: value(chainFormula, tau);
14

15 if |getFormulaAtoms(chainFormula)| > 0 {
16 ProvableTautologies(chainFormula);
17 assert NDDTProof({}, chainFormula);
18 } else {
19 var tau : TruthAssignment := (x: string) => true;
20 ProvableTautologiesSublemma(chainFormula, tau);
21 assert NDDTProof({}, chainFormula);
22 }
23

24 ChainImplicationToAssumptions(sigma, phi);
25 assert NDDTProof(seqToSet(sigma), phi);
26 assert NDDTProof(gamma, phi);
27 }

The special branch for atomless formulae is needed because the tautology lemma

assumes that the formula has at least one atom. If the formula has no atoms, an arbi-

trary truth assignment is sufficient, and the valuation-relative sublemma directly pro-

duces the proof.

4.6 Development Experience with Codex

The derivation-tree formalization was also an experiment in LLM-assisted proof

development. The experience was mixed but informative.

The strongest result occurred at the beginning of the development. After the

DerivationTree datatype was declared and the signatures for the proof predicate and

recursive witness extractor were written, GPT-based Codex was able to draft a com-

plete natural deduction system. This included the validity predicate for the proof con-

structors and the basic extraction functions. The success of this step is likely due to the

close correspondence between the datatype constructors and the natural deduction

rules. Once the representation was fixed, many cases followed a repeated pattern.

Soundness was also well supported. After the signature of NDDTSoundness was

written with the correct precondition and postcondition, Codex drafted the proof and

introduced the additional induction lemma NDDTSoundnessTree. This was the right

structure: instead of trying to prove soundness directly from an existential proof pred-

icate, the proof extracts a tree and performs induction over it. Most node cases were

handled correctly. The main error was the double-negation elimination case, which

required the manually written helper lemma NotNotFEliminationEquivalent.

Completeness was more difficult. The first lemma, semantic deduction, was es-

sentially copied from the previous development. However, when moving to the tautol-

ogy construction, Codex initially tried to use the old height-bounded natural deduction

module. In particular, it attempted to build a proof tree from an NDMaxHeight call and to

call methods from the previous lemma. This was not acceptable in the derivation-tree

development, so the attempt was undone and the prompt was changed to tell Codex

not to use the basic ND module.

After that correction, Codex produced a useful weakening idea. The develop-

ment contains a structural function WeakenTree, together with a proof that weakening

preserves validity.

Listing 4.30: Weakening derivation trees
1 function WeakenTree(dt: DerivationTree, extra: set<Formula>)
2 : DerivationTree
3 decreases dt
4 {
5 match dt
6 case HypothesisNode(gamma, phi’) =>
7 HypothesisNode(gamma + extra, phi’)
8 case AndIntroNode(gamma, phi’, left, right) =>
9 AndIntroNode(gamma + extra, phi’,

10 WeakenTree(left, extra),
11 WeakenTree(right, extra))
12 ...
13 }

This was a productive direction, because weakening is used repeatedly in the

completeness proof. However, Codex’s first attempt at the main tautology sublemma

was not satisfactory. It produced proof code that had the right surface shape but did

not correctly manage the derivation-tree obligations. The sublemma was therefore

manually restructured.

The most effective workflow was the following:

1. write the lemma signature manually;

2. write the difficult representative case manually or semi-manually;

3. ensure that the case uses the correct derivation-tree constructors;

4. ask Codex to complete analogous cases.

This workflow worked especially well after the negation case of the tautology

sublemma was completed. With that example available, Codex completed the impli-

cation, conjunction, and disjunction cases successfully. It also completed the top case

after the bottom case had been drafted manually.

The main limitation observed was representation drift. Codex could recognize

that the derivation-tree proof was analogous to the height-bounded proof, but it some-

times reused the wrong abstraction. In ordinary programming, reusing a nearby helper

may be harmless; in a formalization, using the wrong proof predicate changes the the-

orem being developed. This made it important to keep the specifications precise and

to inspect each generated proof for dependence on the intended module.

Overall, Codex was useful as a proof-engineering assistant but not as an au-

tonomous prover. It was good at generating repetitive proof branches, applying an

already established pattern, and constructing explicit derivation-tree terms once the

intended structure was clear. It was weaker at identifying the crucial invariant or se-

mantic bridge by itself. The development still depended on human decisions about

representation, lemma statements, proof decomposition, and corrections to generated

code.

Chapter 5

Least-Predicate Natural Deduction

This chapter presents a third Dafny formalization of natural deduction, based

on Dafny’s least predicate mechanism. The goal of this version was to represent

derivability directly as an inductively defined predicate, without using either proof

heights or explicit derivation trees.

The previous two formalizations encoded natural deduction in two different ways.

The first used a height-bounded derivability predicate, where proofs were represented

indirectly through the existence of a derivation of bounded height. The second used an

explicit DerivationTree datatype, where each proof was represented as a tree whose

constructors corresponded to natural deduction rules. The least-predicate formaliza-

tion takes a more direct route: the proof relation itself is declared as the least predicate

closed under the rules of natural deduction.

Some names and mathematical definitions were changed in this version to make

the development more compact and better suited to the least-predicate style. Formulae

are represented by the datatype Form rather than Formula, and propositional variables

are represented using the constructor Atom rather than Var. The bottom constant is

named Bot, while Top keeps the same mathematical role as before. Truth assignments

are represented by the type FormEntry, implemented as a set of strings, instead of by

a function from variable names to Booleans. As a result, semantic evaluation is ex-

pressed by the predicate Holds(v, f) rather than by a Boolean-valued function such

as value(f, tau). Semantic consequence is also renamed to Consequence(gamma, psi).

The main proof relation of this formalization is LEP_ND. The intended reading

of LEP_ND(gamma, phi) is that the formula phi is derivable from the assumptions in

gamma. Since LEP_ND is declared as a least predicate, Dafny treats it as the smallest

relation closed under the natural deduction rules written in its definition. This matches

the usual mathematical presentation of natural deduction as an inductively generated

relation.

This representation removes a significant amount of auxiliary proof infrastruc-

ture. In the height-bounded version, proofs had to carry and manipulate numerical

height bounds. In the derivation-tree version, proofs had to construct explicit proof

trees and then show that those trees were valid. In the least-predicate version, the in-

ference rules are written directly as clauses of the derivability predicate. This makes

the formalization shorter and closer to the informal mathematical system.

The methodology was the same as in the derivation-tree chapter. The core defi-

nitions and lemma signatures were provided manually, and Codex was then asked to

draft the proof bodies. In this version, significantly less manual proof-engineering was

needed. Once the least-predicate representation was fixed, Codex was able to draft all

the lemmas. Manual work consisted mainly of writing the specifications, organizing

the modules, and checking that the generated proofs used the intended least-predicate

formalization rather than reverting to one of the earlier encodings.

5.1 Motivation

The motivation for this formalization was to explore whether Dafny’s native sup-

port for inductive predicates could reduce the amount of proof infrastructure required

for natural deduction.

In the derivation-tree formalization, the proof relation was defined by existential

quantification over explicit valid trees:

Listing 5.1: Derivation-tree proof predicate
1 ghost predicate NDDTProof(gamma : set<Formula>, phi : Formula)
2 {
3 exists dt ::
4 NDDTValid(dt) &&
5 RootGamma(dt) == gamma &&
6 RootFormula(dt) == phi
7 }

This representation has the advantage of making proof objects explicit. However,

it also requires several auxiliary definitions: a derivation-tree datatype, root projection

functions, a height function, and a recursive validity predicate.

The least-predicate version removes this layer. Instead of first constructing trees

and then checking their validity, it defines derivability directly.

Listing 5.2: Least-predicate proof relation
1 least predicate LEP_ND[nat](gamma : set<Form>, phi : Form)
2 {
3 ...
4 }

The intended reading is that LEP_ND(gamma, phi) holds exactly when phi is deriv-

able from the assumptions in gamma. Since LEP_ND is declared as a least predicate, Dafny

treats it as the smallest relation closed under the clauses in its body. This is precisely

the expected meaning of an inductively defined proof system.

This representation has two main benefits. First, it avoids explicit proof-tree

bookkeeping. There is no need to maintain root contexts, root formulae, or validity

conditions for proof nodes. Second, it gives soundness a particularly direct structure:

a soundness proof can be written as a least lemma over LEP_ND, and recursive calls

correspond to the smaller derivations appearing in the rule premises.

5.2 The Least-Predicate Proof Relation

The central definition of this chapter is the least predicate LEP_ND. It directly en-

codes the natural deduction rules as disjunctive clauses.

Listing 5.3: Least-predicate natural deduction relation
1 least predicate LEP_ND[nat](gamma : set<Form>, phi : Form)
2 {
3 phi in gamma
4 || phi == Top
5 || (exists phi1, phi2 ::phi == Implies(phi1, phi2)
6 && LEP_ND(gamma + { phi1 }, phi2))
7 || (exists phi1 :: LEP_ND(gamma, phi1)
8 && LEP_ND(gamma, Implies(phi1, phi)))
9 || (exists phi1, phi2 :: phi == And(phi1, phi2)

10 && LEP_ND(gamma, phi1)
11 && LEP_ND(gamma, phi2))
12 // ... the rest of the rules
13 }

For readability and modularity, the formalization also defines separate least pred-

icates for individual rules.

Listing 5.4: Named implication-introduction predicate

1 least predicate LEP_ND_ImpliesIntroduction[nat](gamma : set<Form>, phi : Form)
2 {
3 exists phi1, phi2 ::
4 phi == Implies(phi1, phi2)
5 && LEP_ND(gamma + { phi1 }, phi2)
6 }

These named predicates make it possible to refer to particular rules when needed.

However, the main derivability relation remains LEP_ND.

5.3 Rule Helper Lemmas

A useful feature of the least-predicate encoding is that many rule applications can

be packaged as very small helper lemmas. These helper lemmas expose the natural

deduction rules in a convenient way.

For example, hypothesis introduction is simply the observation that if a formula

belongs to the context, then it is derivable from that context.

Listing 5.5: Hypothesis helper lemma
1 lemma NDHypothesisIntro(gamma : set<Form>, phi : Form)
2 requires phi in gamma
3 ensures LEP_ND(gamma, phi)
4 {
5 assert LEP_ND(gamma, phi) by {
6 assert phi in gamma;
7 }
8 }

Implication introduction is similarly direct.

Listing 5.6: Implication introduction helper lemma
1 lemma NDImpliesIntro(gamma : set<Form>, phi1 : Form, phi2 : Form)
2 requires LEP_ND(gamma + { phi1 }, phi2)
3 ensures LEP_ND(gamma, Implies(phi1, phi2))
4 {
5 assert exists a, b ::
6 Implies(phi1, phi2) == Implies(a, b)
7 && LEP_ND(gamma + { a }, b) by {
8 assert Implies(phi1, phi2) == Implies(phi1, phi2);
9 }

10 assert LEP_ND(gamma, Implies(phi1, phi2));
11 }

Implication elimination is also expressed as a wrapper around the corresponding

clause of LEP_ND.

Listing 5.7: Implication elimination helper lemma
1 lemma NDImpliesElim(gamma : set<Form>, phi1 : Form, phi2 : Form)
2 requires LEP_ND(gamma, phi1)
3 requires LEP_ND(gamma, Implies(phi1, phi2))
4 ensures LEP_ND(gamma, phi2)
5 {
6 assert exists a ::
7 LEP_ND(gamma, a)
8 && LEP_ND(gamma, Implies(a, phi2)) by {
9 assert LEP_ND(gamma, phi1);

10 assert LEP_ND(gamma, Implies(phi1, phi2));
11 }
12 assert LEP_ND(gamma, phi2);
13 }

The same pattern is used for conjunction, disjunction, negation, and bottom elim-

ination. These helper lemmas are not mathematically deep, but they are important for

proof engineering. They make later proofs look like natural deduction derivations

rather than repeated existential-witness assertions over the body of LEP_ND.

Weakening is also much simpler in this formalization than in the derivation-tree

version. In the tree-based formalization, weakening required recursively transforming

proof trees and proving that validity was preserved. In the least-predicate version,

weakening is itself a least lemma.

Listing 5.8: Weakening as a least lemma
1 least lemma Weakening[nat](gamma: set<Form>, delta: set<Form>, f: Form)
2 requires gamma <= delta
3 requires LEP_ND(gamma, f)
4 ensures LEP_ND(delta, f)
5 {
6 }

The empty body is significant. Dafny can prove the lemma from the inductive

structure of LEP_ND. This is one of the clearest examples of the advantage of the least-

predicate encoding: the verification task aligns with Dafny’s built-in induction mech-

anism, so less manual proof infrastructure is needed.

5.4 Soundness

The soundness theorem states that derivability implies semantic consequence.

Listing 5.9: Soundness as a least lemma
1 least lemma {:vcs_split_on_every_assert} Soundness[nat](
2 gamma : set<Form>, phi : Form)

3 requires LEP_ND(gamma, phi)
4 ensures Consequence(gamma, phi)

The proof is written as a least lemma, mirroring the fact that LEP_ND is a least

predicate. This lets Dafny reason by induction over the derivation of LEP_ND(gamma,

phi). The proof body performs case analysis over the possible clauses that could have

produced the derivation.

The first case is the hypothesis rule. If phi in gamma, then any valuation satisfying

all formulae in gamma satisfies phi.

Listing 5.10: Soundness hypothesis case
1 if phi in gamma {
2 assert phi in gamma;
3 }

The top introduction case is also immediate, since Top is true under every valua-

tion.

Listing 5.11: Soundness top case
1 else if phi == Top {
2 forall tau | forall psi :: psi in gamma ==> Holds(tau, psi)
3 ensures Holds(tau, phi)
4 {
5 assert Holds(tau, Top);
6 }
7 }

The implication introduction case illustrates the general structure. If the conclu-

sion is an implication and the consequent is derivable from the context extended with

the antecedent, the induction hypothesis gives semantic consequence for the extended

context. To prove the implication under a valuation satisfying gamma, the proof assumes

that phi1 holds, extends the model with that assumption, and obtains phi2.

Listing 5.12: Soundness implication-introduction case
1 else if exists phi1, phi2 ::
2 phi == Implies(phi1, phi2)
3 && LEP_ND(gamma + { phi1 }, phi2) {
4 var phi1, phi2 :|
5 phi == Implies(phi1, phi2)
6 && LEP_ND(gamma + { phi1 }, phi2);
7

8 Soundness(gamma + { phi1 }, phi2);
9

10 forall tau | forall psi :: psi in gamma ==> Holds(tau, psi)
11 ensures Holds(tau, phi)

12 {
13 if Holds(tau, phi1) {
14 ExtendModel(tau, gamma, phi1);
15 assert Holds(tau, phi2);
16 }
17 }
18 }

The implication elimination case recursively applies soundness to both premises:

the proof of the antecedent and the proof of the implication.

Listing 5.13: Soundness implication-elimination case
1 else if exists phi1 ::
2 LEP_ND(gamma, phi1)
3 && LEP_ND(gamma, Implies(phi1, phi)) {
4 var phi1 :|
5 LEP_ND(gamma, phi1)
6 && LEP_ND(gamma, Implies(phi1, phi));
7

8 Soundness(gamma, phi1);
9 Soundness(gamma, Implies(phi1, phi));

10

11 forall tau | forall psi :: psi in gamma ==> Holds(tau, psi)
12 ensures Holds(tau, phi)
13 {
14 assert Holds(tau, phi1);
15 assert Holds(tau, Implies(phi1, phi));
16 }
17 }

This was also one of the easiest parts for Codex. After the Soundness signature

was written, Codex was able to draft the complete proof. The generated proof followed

the expected structure: split over the clauses of LEP_ND, recursively apply Soundness to

premises, and discharge the semantic consequence goal with a universally quantified

valuation argument.

5.5 Completeness

The completeness theorem states that semantic consequence implies derivability.

Listing 5.14: Completeness theorem signature
1 lemma Completeness(gamma : set<Form>, beta : Form)
2 requires Consequence(gamma, beta)
3 ensures LEP_ND(gamma, beta)

The proof follows the same high-level proof plan used in the previous formaliza-

tions. It consists of three main steps.

First, the assumptions are converted into a sequence and internalized as a chain

of implications.

Second, the resulting formula is shown derivable from the empty context.

Third, the chain of implications is converted back into a derivation from the orig-

inal assumption set.

The helper functions are the same as in the previous developments. A sequence

of assumptions is converted to a nested implication using chainImplication.

Listing 5.15: Chained implication formula
1 function chainImplication(S: seq<Form>, beta: Form): Form
2 decreases |S|
3 {
4 if |S| == 0 then
5 beta
6 else
7 Implies(S[0], chainImplication(S[1..], beta))
8 }

The functions setToSeq and seqToSet mediate between finite sets of assumptions

and the sequence needed by chainImplication.

Listing 5.16: Conversions between sequences and sets
1 function seqToSet(S: seq<Form>): set<Form>
2 decreases |S|
3 {
4 if |S| == 0 then
5 {}
6 else
7 seqToSet(S[1..]) + { S[0] }
8 }
9

10 ghost function setToSeq(S: set<Form>) : seq<Form>
11 decreases |S|
12 {
13 if |S| == 0 then
14 []
15 else
16 var elem :| elem in S;
17 [elem] + setToSeq(S - {elem})
18 }

The top-level completeness proof begins by converting gamma to a sequence sigma

and proving that the conversion preserves the set of assumptions.

Listing 5.17: Set-to-sequence preservation lemma
1 lemma SeqToSetSetToSeq(S : set<Form>)
2 ensures seqToSet(setToSeq(S)) == S

3 decreases |S|

Then the semantic consequence assumption is transferred from gamma to the con-

text represented by sigma.

Listing 5.18: Transferring consequence to the sequence context
1 var sigma := setToSeq(gamma);
2 SeqToSetSetToSeq(gamma);
3

4 assert Consequence(seqToSet(sigma), beta) by {
5 forall tau | forall phi :: phi in seqToSet(sigma) ==> Holds(tau, phi)
6 ensures Holds(tau, beta)
7 {
8 assert forall phi :: phi in gamma ==> Holds(tau, phi);
9 assert Holds(tau, beta);

10 }
11 }

Once this translation is established, the proof calls two major lemmas.

Listing 5.19: Completeness proof outline
1 SemanticDeduction(sigma, beta);
2 ChainImplicationToAssumptions(sigma, beta);
3 assert seqToSet(sigma) == gamma;

The first lemma proves the chained implication from the empty context. The

second lemma converts that proof into a proof of beta from the assumptions in sigma.

5.6 Semantic Deduction and Valid Formulae

The semantic deduction lemma states that if phi follows semantically from the

assumptions in sigma, then the nested implication formed from sigma and phi is deriv-

able from the empty context.

Listing 5.20: Semantic deduction
1 lemma SemanticDeduction(sigma : seq<Form>, phi : Form)
2 requires Consequence(seqToSet(sigma), phi)
3 ensures LEP_ND({}, chainImplication(sigma, phi))
4 {
5 ChainImplicationConsequence(sigma, phi);
6 ValidFormulaIsDerivable(chainImplication(sigma, phi));
7 }

This proof separates semantic reasoning from syntactic derivability. The first

helper lemma proves that the chained implication is semantically valid.

Listing 5.21: Semantic validity of chained implications
1 lemma ChainImplicationConsequence(sigma : seq<Form>, phi : Form)
2 requires Consequence(seqToSet(sigma), phi)
3 ensures Consequence({}, chainImplication(sigma, phi))

Its main recursive component is ChainImplicationHolds, which shows that a val-

uation satisfying the semantic consequence condition also satisfies the chain of impli-

cations.

Listing 5.22: Truth of chained implications
1 lemma ChainImplicationHolds(
2 sigma : seq<Form>, phi : Form, tau : FormEntry)
3 requires ((forall psi :: psi in seqToSet(sigma) ==> Holds(tau, psi))
4 ==> Holds(tau, phi))
5 ensures Holds(tau, chainImplication(sigma, phi))
6 decreases |sigma|

The second helper lemma bridges semantic validity and derivability.

Listing 5.23: Derivability of semantically valid formulae
1 lemma ValidFormulaIsDerivable(phi : Form)
2 requires Consequence({}, phi)
3 ensures LEP_ND({}, phi)
4 {
5 ValidFormulaUnderPartialValuation(phi, {}, getFormulaAtoms(phi), {});
6 assert tauToFormulae({}, {}) == {};
7 }

The proof of ValidFormulaIsDerivable is based on a recursive enumeration of

truth assignments over the atoms of the formula.

The function getFormulaAtoms extracts the set of atoms occurring in a formula.

Listing 5.24: Atoms of a formula
1 function getFormulaAtoms(f : Form) : set<Form>
2 ensures forall a :: a in getFormulaAtoms(f) ==> a.Atom?
3 {
4 match f
5 case Atom(_) => { f }
6 case Implies(a, b) => getFormulaAtoms(a) + getFormulaAtoms(b)
7 case And(a, b) => getFormulaAtoms(a) + getFormulaAtoms(b)
8 case Or(a, b) => getFormulaAtoms(a) + getFormulaAtoms(b)
9 case Not(inner) => getFormulaAtoms(inner)

10 case Top => {}
11 case Bot => {}
12 }

The function tauToFormulae turns a valuation into a set of literals. If an atom

is true under the valuation, the atom itself is included. If it is false, its negation is

included.

Listing 5.25: Valuations as literal contexts
1 function tauToFormulae(atoms : set<Form>, tau : FormEntry) : set<Form>
2 {
3 set x | x in atoms ::
4 match x
5 case Atom(name) =>
6 if Holds(tau, x) then x else Not(x)
7 case Not(Atom(name)) =>
8 if Holds(tau, x) then x else Atom(name)
9 }

The key recursive lemma is ValidFormulaUnderPartialValuation.

Listing 5.26: Recursive valuation enumeration lemma
1 lemma ValidFormulaUnderPartialValuation(
2 phi : Form,
3 assignedAtoms : set<Form>,
4 atomsLeft : set<Form>,
5 tau : FormEntry)
6 requires Consequence({}, phi)
7 requires assignedAtoms + atomsLeft == getFormulaAtoms(phi)
8 requires assignedAtoms * atomsLeft == {}
9 requires forall f :: f in atomsLeft ==> f !in assignedAtoms

10 requires forall f :: f in assignedAtoms + atomsLeft ==> f.Atom?
11 ensures LEP_ND(tauToFormulae(assignedAtoms, tau), phi)
12 decreases |atomsLeft|

If there are no atoms left, the current partial valuation is complete. Since phi

is valid, it holds under this valuation. The proof then invokes the valuation-relative

tautology sublemma.

Listing 5.27: Base case for valuation enumeration
1 if |atomsLeft| == 0 {
2 assert assignedAtoms == getFormulaAtoms(phi);
3 assert Holds(tau, phi);
4 ProvableTautologiesSublemma(phi, tau);
5 assert LEP_ND(tauToFormulae(getFormulaAtoms(phi), tau), phi);
6 }

If atoms remain, the proof chooses one atom and recursively proves phi in both

cases: when the atom is true and when it is false.

Listing 5.28: Branching on one atom
1 var atom :| atom in atomsLeft;
2 var Atom(name) := atom;
3 var remaining := atomsLeft - { atom };
4

5 ValidFormulaUnderPartialValuation(
6 phi, assignedAtoms + { atom }, remaining, tau + { name });
7

8 ValidFormulaUnderPartialValuation(
9 phi, assignedAtoms + { atom }, remaining, tau - { name });

The two recursive calls produce derivations of phi from contexts extended re-

spectively with the atom and with its negation. The formalization then proves the law

of excluded middle and eliminates the disjunction.

Listing 5.29: Combining valuation branches
1 NDLawExcludedMiddle(gamma, atom);
2 NDOrElim(gamma, atom, Not(atom), phi);

This proof has the same mathematical shape as in the derivation-tree formaliza-

tion, but it is simpler to express. There are no explicit OrEliminationNode values to

build, no root contexts to check, and no validity obligations for constructed trees. The

rule helper lemma NDOrElim directly establishes the resulting LEP_ND judgment.

5.7 The Valuation-Relative Tautology Lemma

The central syntactic lemma used in completeness is the valuation-relative tau-

tology sublemma.

Listing 5.30: Valuation-relative tautology lemma
1 lemma ProvableTautologiesSublemma(phi : Form, tau : FormEntry)
2 decreases size(phi)
3 ensures Holds(tau, phi) ==>
4 LEP_ND(tauToFormulae(getFormulaAtoms(phi), tau), phi)
5 ensures !Holds(tau, phi) ==>
6 LEP_ND(tauToFormulae(getFormulaAtoms(phi), tau), Not(phi))

The lemma states that, from the literals corresponding to a valuation, one can

derive either the formula or its negation, depending on whether the formula is true

under that valuation.

For atoms, the proof is immediate because tauToFormulae contains either the

atom or its negation.

Listing 5.31: Atom case of the tautology lemma
1 case Atom(name) => {
2 assert getFormulaAtoms(phi) == { phi };
3 AtomLiteralInTauToFormulae(phi, tau);
4 if Holds(tau, phi) {

5 assert phi in gamma;
6 NDHypothesisIntro(gamma, phi);
7 } else {
8 assert Not(phi) in gamma;
9 NDHypothesisIntro(gamma, Not(phi));

10 }
11 }

For Top, the formula is always derivable by top introduction.

Listing 5.32: Top case of the tautology lemma
1 case Top => {
2 assert Holds(tau, phi);
3 NDTopIntro(gamma);
4 }

For Bot, the formula is never true, so the lemma must prove Not(Bot). This is

handled by deriving Bot from a temporary assumption of Bot, then applying negation

introduction.

Listing 5.33: Deriving not bottom
1 lemma NDNotBot(gamma : set<Form>)
2 ensures LEP_ND(gamma, Not(Bot))
3 {
4 NDHypothesisIntro(gamma + { Bot }, Bot);
5 NDNotIntro(gamma, Bot);
6 }

The negation case is particularly concise in this formalization.

Listing 5.34: Negation case of the tautology lemma
1 case Not(phi1) => {
2 assert getFormulaAtoms(phi) == getFormulaAtoms(phi1);
3 assert Holds(tau, phi) == !Holds(tau, phi1);
4 ProvableTautologiesSublemma(phi1, tau);
5

6 if Holds(tau, phi) {
7 assert !Holds(tau, phi1);
8 assert LEP_ND(gamma, Not(phi1));
9 } else {

10 assert Holds(tau, phi1);
11 assert LEP_ND(gamma, phi1);
12 NDDoubleNegationIntro(gamma, phi1);
13 assert Not(phi) == Not(Not(phi1));
14 }
15 }

Compared with the derivation-tree version, no explicit proof tree for double-

negation introduction is needed. The helper lemma NDDoubleNegationIntro applies

weakening, hypothesis introduction, negation elimination, and negation introduction

directly at the level of LEP_ND.

The implication case follows the usual truth-table reasoning. If the implication

is true because the antecedent is false, the proof derives Not(phi1), assumes phi1, de-

rives contradiction, and then obtains phi2 by bottom elimination before introducing

the implication. If the consequent is true, the proof weakens the derivation of the con-

sequent and introduces the implication. If the implication is false, then the antecedent

is true and the consequent is false; from an assumed implication, implication elimina-

tion gives the consequent, contradicting its negation.

The conjunction and disjunction cases follow analogous patterns. If a conjunction

is true, both conjuncts are derived and conjunction introduction is used. If it is false, the

proof derives the negation of one false conjunct and uses it to refute the conjunction.

If a disjunction is true, one of the disjuncts is derived and disjunction introduction is

used. If it is false, both disjuncts are false; the proof assumes the disjunction, eliminates

it into bottom using two contradictory branches, and introduces the negation of the

disjunction.

This lemma is proof-heavy, but the least-predicate formalization made it substan-

tially easier than the derivation-tree version. Codex was able to draft the full lemma

once the signature and surrounding helper lemmas were available. Very little manual

proof-engineering was needed beyond checking the generated proof and ensuring that

helper lemmas exposed the natural deduction rules in a convenient form.

5.8 Chain Implications Back to Assumptions

The final component of completeness converts a proof of a chained implication

from the empty context into a proof from the corresponding set of assumptions.

Listing 5.35: Chained implications back to assumptions
1 lemma ChainImplicationToAssumptions(sigma: seq<Form>, beta : Form)
2 requires LEP_ND({}, chainImplication(sigma, beta))
3 ensures LEP_ND(seqToSet(sigma), beta)
4 {
5 ChainImplicationToAssumptionsUnder({}, sigma, beta);
6 assert {} + seqToSet(sigma) == seqToSet(sigma);
7 }

The more general helper lemma works under an arbitrary context gamma.

Listing 5.36: Contextual chained-implication helper
1 lemma ChainImplicationToAssumptionsUnder(
2 gamma: set<Form>, sigma: seq<Form>, beta : Form)
3 requires LEP_ND(gamma, chainImplication(sigma, beta))
4 ensures LEP_ND(gamma + seqToSet(sigma), beta)
5 decreases |sigma|

If the sequence is empty, the result is immediate. If the sequence begins with

alpha, the proof weakens the derivation of the implication chain into the larger context

containing all assumptions, derives alpha by hypothesis, and then applies implication

elimination to obtain the shorter chain.

Listing 5.37: One step of chain elimination
1 var alpha := sigma[0];
2 var tail := sigma[1..];
3 var delta := gamma + seqToSet(sigma);
4

5 Weakening(gamma, delta, chainImplication(sigma, beta));
6 assert LEP_ND(delta, Implies(alpha, chainImplication(tail, beta)));
7

8 assert alpha in delta;
9 assert LEP_ND(delta, alpha);

10

11 assert LEP_ND(delta, chainImplication(tail, beta));
12

13 ChainImplicationToAssumptionsUnder(delta, tail, beta);

This proof is considerably shorter than the corresponding derivation-tree proof.

The reason is again that implication elimination is available directly as a clause of

LEP_ND, so the proof does not need to construct an explicit ImpliesEliminationNode.

5.9 Top-Level Completeness

The final completeness theorem combines the previous components.

Listing 5.38: Top-level completeness theorem
1 lemma Completeness(gamma : set<Form>, beta : Form)
2 requires Consequence(gamma, beta)
3 ensures LEP_ND(gamma, beta)
4 {
5 var sigma := setToSeq(gamma);
6 SeqToSetSetToSeq(gamma);
7

8 assert Consequence(seqToSet(sigma), beta);
9

10 SemanticDeduction(sigma, beta);
11 ChainImplicationToAssumptions(sigma, beta);

12

13 assert seqToSet(sigma) == gamma;
14 }

The structure is mathematically the same as in the earlier formalizations:

Γ |= β ⇒|= chainImplication(σ, β) ⇒⊢ chainImplication(σ, β) ⇒ Γ ⊢ β.

The difference is in the representation of the final derivability judgment. In this

chapter, the conclusion is simply:

Listing 5.39: Final derivability judgment
1 LEP_ND(gamma, beta)

No height bound and no explicit proof tree are needed.

Conclusions

This thesis investigated how Dafny can be used to formalize and verify metathe-

oretical results about propositional natural deduction. The goal was not to introduce

new results in logic, but to study how the classical soundness and completeness theo-

rems can be encoded in an auto-active verifier. The development therefore served both

as a formalization case study and as an experiment in proof engineering with Dafny.

The main technical result is that soundness and completeness were verified for

three different representations of natural deduction. The first representation used a

height-bounded judgment, in which derivability is indexed by a natural number. This

approach fits Dafny’s termination requirements well and gives a direct induction prin-

ciple over proof height. However, the height parameter is not part of the ordinary

mathematical statement of natural deduction, so it introduces additional bookkeeping

that must be carried through many lemmas.

The second representation used explicit derivation trees. This style is close to the

informal presentation of natural deduction proof trees, because each constructor corre-

sponds to an inference rule and proof objects can be inspected structurally. At the same

time, the approach requires extra infrastructure for validity of trees, root formulas, con-

texts, weakening, and transformations between trees. As a result, the representation is

conceptually transparent but proof-engineering intensive.

The third representation used Dafny’s least predicate mechanism. This was the

most direct encoding of derivability as an inductively generated relation. It avoided

both artificial height bounds and explicit proof-tree validity predicates, making the

final statements of soundness and completeness closer to the mathematical notation.

The development still required auxiliary lemmas for semantic deduction, tautologies,

implication chains, and context manipulation, but the central proof relation itself was

simpler and more natural.

Across all three formalizations, the most difficult part was not defining propo-

sitional syntax or semantic consequence, but connecting the semantic and syntactic

sides in a form acceptable to Dafny. Mathematically simple facts about sets, sequences,

contexts, and chained implications often had to be stated as separate lemmas. This

illustrates a general feature of auto-active verification: the verifier can check detailed

arguments reliably, but the user must provide enough intermediate structure for the

proof search to succeed.

The thesis also examined the use of large language models as assistants for Dafny

verification. The experiments showed that such tools can be helpful for producing

proof skeletons, filling repetitive cases, proposing helper lemmas, and adapting proof

patterns from one representation to another. They were less reliable for discovering

the main invariants, preserving specifications, and completing large verification argu-

ments without human guidance. In particular, generated proof code had to be checked

carefully for incomplete lemmas, unnecessary assumptions, or changes that altered the

intended theorem.

The overall conclusion is that Dafny is expressive enough to verify substantial

metatheoretical results about propositional natural deduction, but the choice of repre-

sentation strongly affects the amount and kind of proof engineering required. Height-

bounded judgments are convenient for induction and termination, derivation trees

make proof objects explicit, and least predicates provide the cleanest representation of

inductive derivability. Among the three, the least-predicate version appears to be the

closest to the mathematical proof system, while the other two expose useful trade-offs

between verifier control and representational explicitness.

In summary, this thesis shows that auto-active verification can be used not only

for program correctness, but also for machine-checked reasoning about formal logical

systems. It also shows that AI assistants can support this process, provided that they

are used as tools for proof exploration and implementation rather than as substitutes

for formal verification. The final guarantee remains the combination of precise specifi-

cations, human proof design, and Dafny’s verifier.

